Moviéndose de los Modelos Económicos de Equilibrio General Estáticos a los Dinámicos (Notas para un principiante en MPSGE)

Sergey V. Paltsev[*]
Departamento de Economía, Universidad de Colorado[†]

Borrador: [‡]
Junio de 1999 (Revisado en: Junio de 2000) [§]
Versión en Castellano: Enero de 2003

Resumen

Este documento quiere ser una guía para aprendices de MPSGE. Comienza con una breve introducción a la clase de problemas económicos que pueden ser resueltos con MPSGE, y sigue con una descripción detallada de cómo se transforma, paso a paso, un modelo estático simple de equilibrio general en un modelo dinámico del tipo de Ramsey. El modelo está basado en un conjunto simplificado de datos. Se han considerado dos casos: en el primero, la base de datos representa una economía en una senda de crecimiento de estado estacionario; en el segundo, los datos describen una situación en la que la economía no se encuentra en ese estado para el año base. Se incluye código GAMS-MPSGE que puede ser copiado y utilizado como punto de partida para posteriores estudios de modelamiento de dinámica económica.

[*] Agradezco a Thomas Rutherford y a Miles Light por no dejarme hundir cuando me estrellé con el iceberg del MPSGE. Dirección electrónica del autor: paltsev@mit.edu
[†] The current address: 77, Massachussets Ave., E40-429, Massachussets Institute of Technology, Cambridge, MA 02139, USA.
[‡] La última versión de este documento en formato PDF puede ser descargada de http://debreu.colorado.edu/papers/move.pdf
[§] Traducción al castellano por J.C.Segura: Alcaldía Mayor de Bogotá, D.C., Departamento Administrativo de Planeación Distrital. Jsegura@dapd.gov.co url: http://wwwest.uniandes.edu.co/~j-segura

Acerca de la Versión Castellana

Buscando en la Internet material para aprender GAMS/MPSGE tuve la fortuna de hallar este excelente manuscrito de Sergey Paltsev, estudiante (en 2000) del Doctorado en Economía en la Universidad de Colorado en Boulder. Como él mismo advierte en la introducción de su documento, el MPSGE es un poderoso pero pobremente documentado lenguaje porque, tal vez, como Thomas Rutherford —el padre de MPSGE— enfatiza, *el verdadero programador escribe el código, no la documentación*. A falta de las notas de Paltsev, el aprendiz que "se tiene una fe bárbara", pero que cuenta a su vez con una idea muy liberal de lo que es un modelo de equilibrio general computable, en ausencia de un guía paciente se enfrenta al sistema como quien comienza un rompecabezas de diez mil piezas. Con intuición y cierta práctica uno puede calibrar un modelo estático de proporciones reducidas e incluso *correrlo* con éxito, siempre que el objetivo sea aprobar un examen de mitad de semestre. Sin embargo, cuando las dimensiones del modelo aumentan, y el sueldo de economista aprendiz está de por medio, las condiciones de primer orden del problema teórico se tornan difíciles de calibrar con los pocos datos disponibles en el mundo real; el uso de un lenguaje rico y poderoso como GAMS se hace difícil. Es asi que, en tiempos actuales he visto a varios colegas acudir sin éxito (y sin mucha fe la verdad) al venerable GAMS/HERCULES, sin que la aplicación constituya solución a sus problemas.

Aunque el profesor Rutherford proporciona líneas generales de la programación, éstas resultan poco transparentes para el lector común y desprevenido que es el estudiante. Los documentos de Rutherford son de excelente calidad pero omiten muchos detalles que el estudiante raso quisiera conocer: "A Tom no le gusta escribir *papers* para el nivel del principiante", (si bien casi siempre cuenta con algún tiempo para ayudar al estudiante honrado que se encuentra en aprietos). Sobre esto llama la atención Paltsev con gracia y humor: hace que el aprendiz se sienta en el nivel del autor y pierda el temor a errar: *Si Ud. es tan inteligente como Tom, estas notas no son para usted*. Los ejemplos de que Sergey se vale son elementales al comienzo pero al poco de empezar muestran su verdadera sustancia. Y lo mejor de todo es que cuando se termina de leer y seguir los ejemplos, uno quiere ya escribir construir su modelo y escribir su propio *paper* a partir de lo aprendido. Esa es la virtud de este trabajo.

En cuanto a mí, cuando comencé a estudiar este texto, empecé de manera inconsciente a traducirlo. Anotaba en hojas sueltas los contenidos más relevantes a mi juicio. Pronto encontré que había traducido al castellano la mitad del documento. Asi que *puse un e-mail* a Paltsev solicitando su permiso para traducirlo en su totalidad y distribuirlo entre mis compañeros del Magíster en Economía de la Universidad de los Andes de Bogotá, D.C. y entre alumnos de otras facultades. Sergey no solo dio su permiso sino que consideró adecuado incluir ciertas cosas que quizás pudieron haber faltado en la versión original en inglés como, tal la derivación de las FOC del modelo dinámico en su versión NLP. Supongo que será de utilidad a los lectores de habla castellana.

J.C. Segura
Bogotá, D.C., Enero de 2003

1 En lugar de una Introducción

"Este no es un *paper*. Estas son notas de estudio" —Thomas Rutherford, el autor de MPSGE, escribió estas palabras en el primer borrador de este documento. "Yo ya tengo un modelo sobre un modelo de Ramsey. Léalo." —añadió[1]. En esa época era un estudiante graduado pobre con la desesperada necesidad del dinero de una asistencia de investigación. Por eso, no le dije que ya había leído su documento varias veces. Aún después de hacerlo, mi entendimiento acerca de lo que era el MPSGE y sobre cómo construir un modelo dinámico no era mucho mejor que mi entendimiento sobre cómo construir un puente en Nepal. De este modo, reuní aquellas notas como una referencia para mi propio uso. Pensé que si esas notas me habían servido, podrían también ser de utilidad para otros principiantes. La resolución de Tom envió el documento al fondo del cajón de mi escritorio durante un buen tiempo. Estuvo allí hasta cuando repentinamente algunas personas decidieron que yo era poseedor de algún saber especial y me preguntaron: "¿Cómo haces para construir un modelo dinámico en MPSGE?". Obviamente, mi consejo fue que leyeran el documento de Rutherford. Por alguna razón que aún sigue siendo desconocida para mi, casi todos volvieron pidiendo algo más claro y detallado. Mis subsecuentes explicaciones en una mezcla de inglés y ruso los dejaron aún más confundidos. En ese momento decidí que la idea de esas "notas para su propio aprendizaje" no era, después de todo, tan mala. En resumen, si usted es tan inteligente como Tom Rutherford, estas notas no son para Ud. Sin embargo, si se ha sobreestimado un poco y sigue teniendo interrogantes acerca de cómo modelar dinámica en MPSGE, es posible que pueda encontrar las respuestas aquí. A Tom no le gusta escribir documentos a nivel de principiante. Espero ser lo suficientemente ingenuo como para intentarlo, ¿vale? Así pues, vamos a través de lo básico juntos.

En la siguiente sección, trataré de compartir lo que a mi entender es MPSGE y la clase de problemas que pueden ser resueltos con su ayuda. La sección 3 describe cómo un modelo estático sencillo puede ser representado en el formato MPSGE. Si a usted ya le resulta familiar el MPSGE, puede saltarse esta parte e ir directamente a la sección 4, que presenta la transformación de un modelo estático en uno dinámico[2]. Esto para el caso en el cual una economía se encuentra en una senda de estado estacionario inicialmente. La sección 5 presenta los ajustes que es necesario introducir al modelo para representar el caso en el cual la economía no se encuentra en el estado estacionario. Los listados de los programas se proporcionan en los Apéndices. He incluido también Apéndices sobre la conversión MPSGE-MCP y sobre la representación de diferentes formas funcionales en el formato MPSGE. No son directamente relevantes a los modelos dinámicos. Sin embargo, son importantes para una mejor comprensión del proceso de modelaje económico con MPSGE.

Para concluir, siempre me han impresionado esas notas de advertencia a lo FMI y Banco Mundial. Y ahora tengo la oportunidad de poner uno. Aquí va: "Los puntos de vista expresados en este documento son del autor y no necesariamente reflejan los del creador del MPSGE". En un lenguaje elemental, esto significa que estas son sencillamente notas para mi propio aprendizaje. Si usted tiene preguntas acerca del MPSGE, "pregúntele a Tom".

2 MPSGE y los Modelos Económicos de Equilibrio General

El MPSGE (Mathematical Programming System for General Equilibrium Analysis)[*] es un lenguaje de programación diseñado por Thomas Rutherford [1999] a finales de los 80s para resolver modelos de equilibrio económico del tipo Arrow-Debreu. El MPSGE es un poderoso pero pobremente documentado lenguaje. Gracias a este hecho, la receta usual para dominar el MPSGE es el *learning by doing*.

---

[1] Me referiré a este paper sobre modelamiento dinámico (Lau, Pahlke, Rutherford [1977]) como el documento de Rutherford. Está disponible en: http:/nash.colorado.edu/tomruth/primer/paper.htm

[2] La discusión de este documento se limita al modelo de Ramsey. Otra aproximación a la introducción de la dinámica es la de los modelos de generaciones traslapadas (OLG). Para la formulación de modelos OLG en MPSGE, ver Rutherford [1998].

[*] Sistema de Programación Matemática para el Análisis de Equilibrio General [N. De T.]

MPSGE utiliza el lenguaje de programación GAMS (General Algebraic Modeling System) como interfase[3]. Como consecuencia, si quiere utilizar MPSGE, necesita también aprender la sintaxis GAMS[4] (Brooke, Kendrick, Meeraus [1992]). Según se deriva de su nombre, MPSGE resuelve modelos económicos de equilibrio general computables (CGE). Estos modelos comprenden agentes económicos que interactúan entre ellos. La palabra "General" significa que *todos* los flujos económicos son contabilizados, esto es, existe un *uso* para cada *fuente*. Encontrar un equilibrio para un modelo CGE implica encontrar precios, cantidades e ingresos de equilibrio.

Como ilustración de un modelo económico básico, considere la tarea de hallar un equilibrio en una economía (que llamaré *País de las Maravillas*), que consta de dos agentes económicos: *consumidores* y *productores*, Los consumidores tienen una dotación inicial de trabajo L y de capital K. En aras de sencillez, existe un único consumidor representativo[5] *CONS* en el País de las Maravillas. El consumidor deriva su ingreso I de las ventas de sus dotaciones. Así, obtiene la cesta de bienes de consumo de su elección. En la economía se dispone de dos bienes, X e Y. El consumidor obtiene utilidad del consumo de esos bienes. Los productores son las firmas que toman las dotaciones iniciales del consumidor como insumos y las convierten en productos. Los dos sectores productivos, X e Y están caracterizados por las tecnologías disponibles, F y G respectivamente.

Queremos determinar los precios y las cantidades que hacen máximos los beneficios del productor y la utilidad del consumidor[6]. La solución de este problema (que es un problema simple de Arrow-Debreu [1954]) puede encontrarse utilizando una formulación no lineal. El problema puede ser representado como uno de optimización del consumidor, sujeto a restricciones de ingresos, tecnología y viabilidad.

$$\max\ W(X,Y) \quad \text{s.a.} \quad \begin{aligned} I &= p_x X + p_y Y \\ X &= F(K_x, L_x) \\ Y &= G(K_y, L_y) \\ L &= L_x + L_y \\ K &= K_x + K_y \end{aligned}$$

donde W es una función de utilidad, p_x y p_y son los precios de los bienes X e Y; K_x y L_x son el capital y el trabajo empleados en la producción del sector X, K_y y L_y son el capital y el trabajo empleados en el sector Y. Este es un problema de optimización propio de textos estándar de microeconomía y una técnica usual para encontrar la solución es el método de los multiplicadores de Lagrange. Este problema puede ser resuelto con GAMS como un programa no lineal (NLP).

Existen algunos casos (como la presencia de diferentes consumidores, impuestos u otras distorsiones) que no permiten resolver el problema de hallar un equilibrio de mercado como un problema de optimización. Así, el problema debe ser aproximado en una forma diferente. Puede ser convertido en un problema mixto de complementariedad (MCP) y ser resuelto como un sistema de ecuaciones no lineales. Los problemas NLP constituyen un subconjunto de los MCP y MPSGE encuentra el equilibrio como una solución a un MCP.

He aquí la primera prueba para verificar si usted debe seguir leyendo estas notas o si más bien debería dirigirse directamente al documento de Rutherford [1997]. Nunca antes había escuchado hablar de la formulación MCP. Cuando tuve a bien preguntar, una *buena* persona me dijo que era muy sencillo, algo así como:

$$\text{Dada}: \quad f : R^n \to R^n$$
$$\text{Encontrar}: \quad z \in R^n$$
$$\text{s.a.} \quad f(z) \geq 0, z \geq 0, \ z^T f(z) = 0.$$

[3] Para aprender más acerca de GAMS, visite http://www.gams.com

[4] Hacia abril de 2000, una versión estudiantil gratuita de GAMS puede ser descargada de http://www.gams.de/5download/cd.htm

[5] O, en forma equivalente, una población de hogares idénticos.

[6] A los no economistas puede resultar divertido eso de que existe un solo consumidor que posee las firmas y compra los bienes producidos por aquellas. Sin embargo, para los economistas, con su "forma económica de pensar" esto es natural. El significado de este supuesto es que se ignoran consideraciones distribucionales entre los diferentes tipos de consumidores.

Si en la proposición anterior todo le resulta claro, Ud. está perdiendo su tiempo con estas notas. Guárdelas y vuelva al trabajo real!

Si sigue allí, tratemos entonces de convertir los símbolos matemáticos en un idioma sencillo. Debo advertir que esto no ayudará mucho al principiante en razón de que suena algo así como: dada la función f entre dos conjuntos n-dimensionales de números reales (la función f asigna a cada miembro del primer conjunto exactamente un miembro del segundo), encontrar z que pertenece a un conjunto n-dimensional de números reales, tal que la función $f(z)$ sea mayor o igual que cero, z sea mayor o igual a cero y que la condición de holgura complementaria[7] asociada se satisfaga. La condición de holgura complementaria requiere ya sea que z sea igual a cero (esto es, el multiplicador dual se desvanece), o que la restricción de desigualdad se satisfaga con igualdad estricta, o ambas.

Sin embargo, estas palabras son sencillamente una forma científica de ocultar algo simple. ¿recuerdan de sus matemáticas de octavo grado la solución de una ecuación como: $x(5 - x) = 0$? Si, esta es $x = 0$ y $x = 5$. Para hacer más clara la comparación, la ecuación puede ser rescrita como $xf(x) = 0$, donde $f(x) = 5 - x$. Esto lleva a la condición de que x o que $f(x)$ tengan que ser igual a cero. Esta es la idea principal detrás de la MCP.

En el caso en el que no tenemos una única x sino un vector $\bar{x} = \left(x_1, x_2, \cdots, x_n\right)$, existirá un sistema de n ecuaciones del tipo $\bar{x}f\left(\bar{x}\right) = 0$ que conforman un problema MCP. La palabra "mixto" en MCP refleja el hecho de que la solución es una mezcla de igualdades $f(x) = 0$ y de desigualdades $f(x) > 0$.

Mathiesen [1985] demostró que el modelo de equilibrio económico de Arrow-Debreu puede ser representado como un MCP, en el cual deben satisfacerse tres tipos de desigualdades: una condición de beneficio cero, una condición de vaciado de mercado y una condición de balance de ingresos. En la solución de un problema MCP se involucran tres variables no negativas: precios, cantidades (que en MPSGE se denominan *niveles de actividad*) y niveles de ingresos.

La *condición de beneficio cero* requiere que cualquier actividad que opera a intensidad positiva debe generar beneficios puros nulos (es decir, el valor de los insumos debe ser mayor o igual que el valor de los productos). Los niveles de actividad y para sectores de rendimientos constantes a escala son las variables asociadas a esta condición. Esto implica que $y > 0$ (se produce una cantidad positiva de y) y el *beneficio* es cero o bien que el *beneficio* es negativo y $y = 0$ (ningún nivel de producción toma lugar). En términos de MCP, la siguiente condición debe satisfacerse para cada sector en una economía:

$$—\textit{beneficio} \geq 0,\ y \geq 0,\ \textit{producto}^T(—\textit{beneficio}) = 0$$

La *condición de vaciado de mercado* requiere que cualquier bien con precio positivo debe exhibir un balance entre su oferta y su demanda y que cualquier bien con exceso de oferta debe mostrar precio cero. El vector de precios p (que incluye los precios de los bienes finales, de los bienes intermedios y de los factores de producción) es la variable asociada. En términos de MCP, la siguiente condición debe ser satisfecha para cada bien y cada factor de producción:

$$\textit{Oferta} - \textit{Demanda} \geq 0,\quad p \geq 0,\quad p^T(\textit{Oferta} - \textit{Demanda}) = 0$$

La *condición de balance de ingresos* requiere que para cada agente (incluyendo entidades gubernamentales) el valor del ingreso iguale al valor de las dotaciones. Es posible formular esta condición den términos de MCP:

$$(\textit{dotación - ingreso}) \geq 0,\ \textit{ingreso} \geq 0,\ \textit{ingreso}^T(\textit{dotación-ingreso}) = 0$$

En cualquier caso, el ingreso usualmente no es igual a cero y la condición *dotación-ingreso* = 0 se satisface como igualdad. Como tal, el balance de ingresos es más definicional que de complementariedad.

[7] Una expresión escrita como $x^Ty=0$ significa que $x_iy_i = 0$, para todo i=1,..., n. Las variables x_i y y_i se denominan *par complementario* y se dice que son *complemento* una de otra.

A fin de ilustrar las condiciones de equilibrio, considere el ejemplo del País de las Maravillas. Por razones que serán obvias más adelante, se introducirá el sector productivo *W que representará la utilidad derivada del consumo de X* y de *Y*. De manera paralela, P_w es el precio de la producción del sector *W*. En primer lugar, considere las condiciones de beneficio cero. En el ejemplo del País de las Maravillas, un nivel de actividad *y* es un vector con los siguientes componentes $y = (X,Y,W)$. El beneficio en un sector particular está definido como la diferencia entre el ingreso total y el costo total. Asumiremos que los niveles de actividad X,Y y W son positivos (o sea, la condición $y \geq 0$ se cumple con desigualdad estricta $y > 0$). De esta forma, las condiciones de beneficio cero ((—*beneficio*) = 0) pueden escribirse como:

$$—(P_x - C_x(w,r)) = 0 \tag{1}$$

$$—(P_y - C_y(w,r)) = 0 \tag{2}$$

$$—(P_w - e(P_x,P_y)) = 0 \tag{3}$$

donde C_x y C_y son las funciones de costo unitario correspondientes a X e Y (el costo de producir una unidad de un bien a los precios *w* y *r* de los factores); *e* es la función de costo (gasto) unitario de *W* (el costo de comprar una unidad de utilidad a los precios P_x y P_y).

La condición de vaciado de mercado requiere que, en presencia de precios positivos, las ofertas deben ser iguales a las demandas. Esto se representa mediante las siguientes ecuaciones

$$X = \frac{\partial e}{\partial P_x} W \tag{4}$$

$$Y = \frac{\partial e}{\partial P_y} W \tag{5}$$

$$W = \frac{I}{P_w} \tag{6}$$

$$L = \frac{\partial C_x}{\partial w} X + \frac{\partial C_y}{\partial w} Y \tag{7}$$

$$K = \frac{\partial C_x}{\partial r} X + \frac{\partial C_y}{\partial r} Y \tag{8}$$

En el lado de la demanda, las derivadas parciales $\partial e/\partial P_i$, $\partial C_i/\partial w$, *y* $\partial C_i/\partial r$ representan el Lema de Shephard[8].

La condición de balance de ingresos establece que el valor de las dotaciones debe igualar el ingreso del consumidor:

$$wL + rK = I \tag{9}$$

MPSGE resolverá nueve ecuaciones (1-9) para nueve incógnitas: niveles de actividad, *X*,*Y*,*W*; precios, P_x, P_y, P_w, *w* y *r*; y el ingreso *I*. Debe notarse que las condiciones de equilibrio son difíciles de formular en forma explícita en ciertos problemas. Las buenas noticias para un usuario de MPSGE consisten en que uno no necesita gastar tiempo en la derivación de las condiciones de equilibrio. MPSGE las construye automáticamente. Sin embargo, siempre es útil entender lo que la representación de un modelo MPSGE

[8] Para mas información sobre el Lema de Shephard ver, por ejemplo, Varian [1992].

particular representa[9]. A efectos de comparación, en el Apéndice 2 se presentan formulaciones alternativas de un modelo sencillo de reforma fiscal y comercio internacional (Shoven y Whalley [1984]) tanto en MPSGE como en su forma algebraica (formulación MCP). En el documento de Rutherford es posible hallar una representación MCP del modelo de Ramsey. A fin de correr un programa MPSGE, se necesita de *solvers* MCP. GAMS incluye dos de estos: MILES y PATH. El palabras de Tom, MILES es su niño abandonado, lo cual deja a PATH como primera opción.

3 Un Modelo Estático

En la sección anterior se ha discutido brevemente la manera como MPSGE encuentra un equilibrio. Es bueno saber que MPSGE lo hará no como un problema de optimización sino, más bien mediante el hallazgo de un equilibrio en un sistema de inecuaciones. En cualquier caso, al crear un programa MPSGE sencillo el usuario no necesita saber esto. Para el principiante, lo que es de mayor importancia es la sintaxis del lenguaje MPSGE[10]. Una gran ventaja de MPSGE es que está basado en funciones anidadas de utilidad y de producción de elasticidad de sustitución constante (CES)[11]. El uso de funciones anidadas puede proporcionar una representación flexible de cómo las entradas se relacionan. Para introducir la sintaxis MPSGE[12], consideraremos de nuevo el ejemplo del País de las Maravillas descrito en la sección 2. El programa GAMS-MPSGE del modelo estático del País de las Maravillas (`move1_1.gms`) se registra en el Apéndice 3.

Los modelos económicos se basan en datos que a su turno se organizan de manera usual en Matrices de Contabilidad Social (SAM)[13]. Asumiremos la siguiente información, representada en la Tabla 1 en la cual la SAM tiene tres sectores productivos X, Y y W (se declaran en el bloque `$SECTORS` del programa MPSGE), un consumidor (declarado en el bloque `$CONSUMERS`), y cinco mercados X, Y, W, L y K (declarados en el bloque `$COMMODITIES`). Dado que las variables asociadas a los bienes son sus precios, usualmente empiezan con la letra P.

	Sectores Productivos			Consumidores
Mercados	X	Y	W	CONS
PX	100		-100	
PY		100	-100	
PW			200	-200
PL	-40	-60		100
PK	-60	-40		100

Tabla 1. Una SAM para un modelo estático (ejemplo del País de las Maravillas)

Una SAM describe un conjunto de datos usualmente conocido como base de *SAM de benchmark*. Los números representan valores (precios por cantidades) de las transacciones económicas en momento determinado. Una entrada positiva representa el valor de la producción (o las ventas). Una entrada negativa refleja el valor de un insumo (o las compras). Las condiciones de beneficio cero, de vaciado de mercado y de balance de ingresos implican suma cero a lo largo de filas y columnas.

Para ilustrar como se balancea la SAM, considere las siguientes filas y columna de la Tabla 1. El sector productivo X adquiere 40 unidades[14] de trabajo y 60 de capital y vende 100 unidades del bien X. La suma a lo largo de la columna X es igual a cero. Esto muestra que no existen excesos de beneficios (beneficio cero) en el sector X. La fila PX muestra que las 100 unidades de X producidas y vendidas (la entrada positiva

[9] La primera vez que vi un código MPSGE me dije: "necesito ver el álgebra subyacente". Posteriormente comprendí que la representación MPSGE es mucho más atractiva e intuitiva que la formulación algebraica. Así, ahora, aquella buena persona siempre sonrie cuando digo "No quiero ver esas feas ecuaciones MCP, muéstreme un código de MPSGE".

[10] La sintaxis del MPSGE puede encontrarse en http://debreu.colorado.edu/mainpage/mpsge.htm

[11] Vea el Apéndice 1 para ejemplos de conversón de diferentes funciones al formato MPSGE.

[12] En el tratamiento de modelos simples estáticos y dinámicos, muchas característicos del MPSGE no serán discutidas. Para práctica adicional, vea los ejemplos de Mrjusen (Markusen y Rutherford [1995]), disponibles en: http://nash.colorado.edu./tomruth/6433/markusen.htm.

[13] Para más información sobre cómo convertir datos de insumo-producto en modleos MPSGE, ver Rutherford y Paltsev [1999].

[14] Asegúrese de recordar que las unidades representan aquí el valor, esto es, precios por cantidades.

de 100) son consumidas por el sector W (la entrada negativa de 100). La suma a lo largo de la fila PX es igual a cero. Esto representa la igualdad entre oferta y demanda (vaciado de mercado) para el bien X.

Por sencillez, algunas veces se adopta la convención de Harberger en una SAM de benchmark. Esta consiste en normalizar todos los precios a 1. Así, las cantidades en la SAM representan gastos, o cuanto de un bien o factor se puede comprar con $1. Debe notarse que una economía Arrow-Debreu depende únicamente de los precios relativos. Al doblar todos los precios se doblan los beneficios monetarios y el ingreso, lo cual resulta en la misma solución para las cantidades (o niveles de actividad). Los precios absolutos no tienen impacto en el equilibrio resultante.

Como ya se habrá notado, MPSGE no requiere que el usuario formule la representación algebraica de las funciones de producción y utilidad. El usuario solo necesita proporcionar las cantidades de referencia, los precios de referencia y las elasticidades. Basado en la información provista, MPSGE construye las funciones de producción y de utilidad subyacentes. En la solución, MPSGE retorna los valores de equilibrio para las variables, descritas en `$SECTORS, $COMMODITIES, y $CONSUMERS.`

En el ejemplo del País de las Maravillas, existen dos productos (X y Y), dos factores (L y K) y un consumidor (CONS). Una fila y columna extras (PW y W) se incorporan para representar explícitamente la utilidad derivada del consumo agregado. Como discutiremos posteriormente, esto pudo haberse hecho directamente en el bloque `$DEMAND`[15].

El programa empieza con una declaración del título que habrá de ser impreso en el listado de la solución. Usualmente sigue la declaración de escalares, parámetros y conjuntos en el formato GAMS. No necesitamos incluirlos en nuestro modelo estático simple. La siguiente porción está en formato MPSGE.

```
$ONTEXT
$MODEL:NAME
```

Estos dos bloques conmutan el control del GAMS al compilador del MPSGE. El usuario escoge el nombre del modelo. Por ejemplo, nuestro modelo se llama MOVE1_1. El nombre del modelo debe ser un nombre de archivo legítimo puesto que se generará un archivo NAME.GEN (MOVE1_1.GEN).

```
$SECTORS:
```

En este bloque se declaran los sectores productivos. Determinan como los insumos son convertidos en productos. Aquí, las variables son niveles de actividad y están asociadas a las condiciones de beneficio cero (asegúrese de que un sector productivo obtiene beneficios cero). En el modelo del país de las maravillas, tenemos tres sectores: X, Y, y W. Es posible hacer comentarios en la misma linea luego de la marca !.

```
$COMMODITIES:
```

Los bienes se declaran en este bloque. Las variables aquí son los precios. Están asociados a las condiciones de vaciado de mercado (es preciso garantizar que las ofertas son iguales a las demandas). Tenemos cinco bienes: dos precios de productos PX, PY, dos precios de factores, PK, PL y un índice de precios PW para el bienestar (es decir, la utilidad derivada del consumo agregado)

```
$CONSUMERS:
```

Aquí se describen los consumidores que ofrecen los factores y perciben utilidad. La variable aquí es el ingreso proveniente de todas las fuentes. Está asociado con la condición de balance de ingresos. (hay que garantizar que el ingreso total es igual al consumo total (o demanda total)). En nuestro caso tenemos un único consumidor representativo, CONS.

```
$PROD:
```

El bloque de producción construye la función subyacente de producción (la regla mediante la cual los insumos son convertidos en productos). Los sectores y los bienes utilizados en un bloque de producción deben ser declarados en los bloques respectivos. Considere el primer bloque de producción, para X (los bloques de producción para Y y W tienen una estructura similar):

```
$PROD:x  s:1
```

[15] Sin embargo, un bloque separado para la utilidad resulta útil en la introducción de un impuesto sobre el consumo y para el análisis de bienestar.

```
      O:PX  Q:100
      I:PL  Q:40
      IPK   Q:60
```

Este bloque describe una función de producción Cobb-Douglas (lo podemos ver gracias a `s:1` que significa que la elasticidad de sustitución entre factores es igual a 1). Los insumos aquí son PL y PK (que podemos observar de `I:PL` y `I:PK`). El producto es PX (`O:PX`). Este sector productivo convierte 40 unidades de PL y 60 unidades de PK en 100 unidades de PX. Matemáticamente, la función de producción para esta tecnología puede ser escrita como $X = \phi L^{0.4} K^{0.6}$ (Vea el Apéndice 1 para la conversión de funciones de producción y utilidad en el formato MPSGE). La figura 1 ilustra la calibración de una función de producción a los precios y cantidades de referencia.



Fig.1. Calibración de la Función de Producción

MPSGE especifica la función de producción con un solo punto de referencia. En este ejemplo hemos especificado explícitamente solamente las cantidades. Los precios de referencia se fijan en 1 por defecto. Los precios relativos de los insumos determinan la pendiente de la isocosto (que a su vez es igual a la pendiente de la isocuanta en las cantidades de referencia). Esto significa que nuestro bloque de producción es idéntico a:

```
$PROD:X s:1
      O:PX  Q:100 P:1
      I:PL  Q:40  P:1
      I:PK  Q:60  P:1
```

La curvatura de la isocuanta está determinada por *s*, la elasticidad de sustitución entre insumos (*s*:1 corresponde a una función de producción Cobb-Douglas). El valor por defecto de la elasticidad es cero. Las cantidades y precios de referencia son utilizados únicamente en la calibración. *El solucionador no los utiliza como valores iniciales de las variables.* El bloque de producción Y es similar al bloque de producción de X.

El bloque de producción de W (Bienestar) sirve como herramienta de conversión de los bienes X e Y en utilidad derivada del consumo agregado. En este modelo simple hemos introducido este bloque de producción a efectos de claridad únicamente. Podríamos haber eliminado el bloque de producción W y haber cambiado el de demanda en[16]

```
$DEMAND:CONS
      D:PX  Q:100
      D:PY  Q:100
      E:PL  Q:100
      E:PK  Q:100
```

De cualquier manera, en modelos más complicados, la representación del bienestar como un bloque de producción puede resultar de utilidad (por ejemplo en el análisis de un impuesto al consumo).

`$DEMAND:`
El bloque de demanda representa las restricciones de balance de ingresos.

```
$DEMAND:CONS
      D:PW  Q:200
      E:PL  Q:100
      E:PK  Q:100
```

Un consumidor recibe ingresos de sus dotaciones de L y K (`E:PK` y `E:PL`) y demanda PW bienes (`D:PW`). Las cantidades de referencia (los campos `Q`) se utilizan en la calibración de la función de utilidad de la misma forma que el bloque de producción calibra la función de producción.

```
$OFFTEXT
$SYSINCLUDE mpsgeset NAME
```

Estos dos bloques hacen que MPSGE devuelva el control a GAMS.

`NAME.ITERLIM = 0;`
Este comando determina el número de iteraciones. Con el límite de iteraciones en cero, se le pide al solucionador no que resuelva el modelo sino que retorne los valores en los cuales los datos iniciales están basados. Es importante fijar el límite de iteraciones en cero es importante al asegurar que la información de referencia (la información descrita en una SAM) represente una solución de equilibrio.

```
$INCLUDE NAME.GEN
SOLVE NAME USING MCP;
```
Estos dos bloques se utilizan para correr el modelo.

La solución del modelo estático (reportado en un archivo de listado (`move1_1.lst`)) es:

```
          LOWER          LEVEL          UPPER          MARGINAL

VAR X       .            1.0000         +INF              .
VAR Y       .            1.0000         +INF              .
VAR W       .            1.0000         +INF              .
VAR PX      .            1.0000         +INF              .
VAR PY      .            1.0000         +INF              .
VAR PL      .            1.0000         +INF              .
VAR PK      .            1.0000         +INF              .
VAR PW      .            1.0000         +INF              .
VAR CONS    .          200.0000         +INF              .
```

En la solución, las columnas LOWER y UPPER muestran los límites de las variables. El cero se representa con ".". INF significa infinito. La columna LEVEL reporta el valor de solución. La columna

[16] Vea el código del programa `move1_2.gms` en el Apéndice 3. Es posible correrlo para comparar los resultados `move1_1.gms`.

MARGINAL muestra el valor de la variable de holgura (sombra) complementaria. Es preciso prestar suficiente atención al las columnas LEVE y MARGINAL. La holgura complementaria implica que en equilibrio el valor de la variable será positivo o que el valor del marginal lo sea, pero no ambos. Si los dos son positivos, necesitamos chequear nuestro modelo.

LA solución de nuestro modelo estático nos dice que la información de referencia (representada en la SAM) es consistente con el equilibrio del modelo. Podemos verificar esto porque todos los marginales son iguales a cero luego de la réplica del la referencia (es decir, luego de resolver el modelo con ITERLIM=0). Los niveles de actividad de equilibrio X, Y y W son iguales a 1 (*y no a 100,100 y 200 como en la SAM!*). Los precios de equilibrio son iguales a 1. El nivel de ingreso de equilibrio del consumidor representativo es igual a 200.

Es algo confuso para los que comienzan el que los niveles de actividad en la solución sean todos iguales a 1. Esto es justamente debido al reescalamiento y se hace con el propósito de análisis futuros de experimentos contrafactuales en los cuales el usuario puede ver fácilmente el porcentaje de cambio de un valor desviado de 1. Si deseamos ver los niveles de actividad sin escalar, necesitamos multiplicar los niveles de la solución por los campos `Q:` en los bloques de producción. Por ejemplo, para el nivel de actividad X, esto es igual a 1*100=100.

Hay otras maneras de obtener los niveles de actividad actuales en el listado de solución. Una forma posible se reescalar los bloques de producción y asignar valores iniciales de las actividades según se hace en el programa `move1_3.gms`. La solución en este caso replica directamente los datos de referencia:

```
              LOWER          LEVEL          UPPER         MARGINAL

VAR X           .           100.0000        +INF             .
VAR Y           .           100.0000        +INF             .
VAR W           .           200.0000        +INF             .
VAR PX          .             1.0000        +INF             .
VAR PY          .             1.0000        +INF             .
VAR PL          .             1.0000        +INF             .
VAR PK          .             1.0000        +INF             .
VAR PW          .             1.0000        +INF             .
VAR CONS        .           200.0000        +INF             .
```

En cualquier caso, por razones numéricas, es aconsejable escalar los valores de equilibrio a números cercanos a la unidad. Es por esto que es mejor utilizar un bloque REPORT como se ilustra en el programa `move1_4.gms`, en el cual la SAM inicial se recrea utilizando los valores reportados.

Todos los valores iniciales de los niveles de actividad y de los precios son iguales a 1 por defecto. La variable CONS, que representa el nivel de ingreso de un consumidor es más bien diferente. Esta se encuentra determinada por las dotaciones. *Recuerde de la condición de balance de ingresos que el ingreso es igual al valor total de las dotaciones y, a su turno, igual a la demanda total.* En el caso del modelo estático es igual a 200 porque la dotación de capital (`E:PK`) es 100, la dotación de trabajo (`E:PL`) es 100, y los precios, PK y PL son iguales a 1 por defecto.

Como se habrá notado, el modelo solo determina precios relativos. Así, si asignamos a todos los precios iniciales el valor 2, la variable CONS será calculada de manera que sea igual a 400. Podemos chequear esto modificando el modelo estático. Los niveles iniciales de precios pueden ser asignados luego de `$SYSINCLUDE mpsgeset NAME` pero antes del comando `$INCLUDE NAME.GEN` de la siguiente manera:

```
PX.L = 2;
PY.L = 2;
PW.L = 2;
PK.L = 2;
PL.L = 2;
```

De esta forma el listado de solución mostrará un incremento en todos los precios y en el nivel de ingreso.

```
LOWER          LEVEL          UPPER          MARGINAL

VAR X          .              1.0000         +INF           .
VAR Y          .              1.0000         +INF           .
VAR W          .              1.0000         +INF           .
VAR PX         .              2.0000         +INF           .
VAR PY         .              2.0000         +INF           .
VAR PL         .              2.0000         +INF           .
VAR PK         .              2.0000         +INF           .
VAR PW         .              2.0000         +INF           .
VAR CONS       .            400.0000         +INF           .
```

Gracias a que solo se determinan precios relativos, usualmente uno de los precios es fijado por el usuario. Si ningún precio se predetermina fijo inicialmente (esto es, ningún bien se escoge como numerario), el MPSGE hará una normalización. En el caso de nuestro modelo estático, el nivel de ingreso del consumidor CONS es el que fija. Esto puede observarse en el listado de solución (`.lst`) al mirar el mensaje `Default price normalization using income for CONS`.

El modelo puede también ser especificado en un formato diferente. En lugar de usar números en los campos de precios y cantidades, pudimos haber introducido parámetros al comienzo de programa y utilizarlos para la calibración en los campos de los bloques de producción y de demanda. Un ejemplo de esta sintaxis vectorial se presenta en el programa `move1_5.gms`.

Luego de la calibración de la referencia, se efectúan experimentos contrafactuales (por ejemplo, la introducción de impuestos o cambios en la dotación de trabajo). Es importante recordar cambiar el límite de las iteraciones antes de correr los contrafactuales, lo cual se logra con un comando GAMS como `NAME.ITERLIM=2000;`. El solucionador se detendrá tan pronto como el límite se haya alcanzado, si no se ha encontrado solución antes del número de iteraciones definidas por el usuario. En este documento, no describiremos ningún contrafactual con nuestro modelo estático[17]. Nuestro interés aquí es diferente y consiste en convertir un modelo estático en uno dinámico.

4 Modelos Dinámicos

Cuando empecé a aprender modelaje económico inmediatamente deseé construir un modelo dinámico. Quería predecir el futuro pues, ¿qué clase de predicciones uno puede hacer con un modelo estático? Posteriormente entendí que mis intentos de modelar dinámica eran como desear construir una nave espacial con un destornillador. De hecho esperaba demasiado de los modelos dinámicos si bien su potencial está muy lejos de predecir el futuro. Un modelo puede decirle, más o menos, lo que va a pasar si no hay choques ni cambios estructurales en la economía. En adición, uno requiere hacer supuestos acerca de la tasa de crecimiento económico en varias décadas en el futuro, la tasa de preferencia intertemporal, la tasa de crecimiento poblacional, la inflación, la depreciación, etc. Todos estos supuestos necesarios nos llevan bien aparte de la realidad. Pero aquellos que hacen la política necesitan tomar decisiones así como los economistas necesitan proporcionar respuestas acerca del futuro. Como consecuencia, los modelos de equilibrio general dinámicos son importantes herramientas para la evaluación de la política pública. Pero ahora se que es mucho mas informativo desarrollar un buen modelo estático que construir uno dinámico pobre (y hasta ahora, nunca he visto un buen modelo dinámico).

Con esta motivación, intentaremos desarrollar un modelo dinámico basado en el ejemplo del País de las Maravillas. Existen diferentes aproximaciones al modelaje económico dinámico (ver Ginsburgh y Keyzer [1997] para una panorámica). Limitaremos nuestra discusión al modelo simple de Ramsey. El modelo se comporta de manera diferente si se encuentra o no en lo que se llama *estado estacionario*. El *estado estacionario* se define como una situación en la cual las diferentes cantidades (capital, producto, consumo, etc.) crecen a tasas constantes. Empezaremos nuestras consideraciones con una situación en la cual la información de referencia describe una economía en estado estacionario en el período base.

[17] Vea los ejemplos de Markusen para préctivar (Markusen y Rutherford [1995]).

4.1. Una economía en estado estacionario en el período base

En la versión dinámica del ejemplo del País de las Maravillas, el consumidor representativo maximiza el valor presente de la utilidad de su plazo de vida

$$max \sum_{t=0}^{\infty} \left(\frac{1}{1+\rho} \right)^t U(c_t) \tag{10}$$

donde t son los períodos de tiempo, ρ es el factor intertemporal de descuento, U es la función de utilidad, y c_t es el consumo del período t. El consumidor enfrenta diversas restricciones. Primero, el producto total de la economía se divide en consumo e inversión, I_t. Segundo, el capital se deprecia a la tasa δ. Tercero, la inversión no puede ser negativa. Estas restricciones pueden escribirse de la siguiente forma:

$$c_t \leq F(K_t, L_t) - I_t \tag{11}$$

$$K_{t+1} = K_t(1-\delta) + I_t \tag{12}$$

$$I_t \geq 0 \tag{13}$$

donde K es el capital y F representa la función de producción. Al resolver el problema de maximización de la utilidad obtenemos las siguientes condiciones de primer orden:

$$P_t = \left(\frac{1}{1+\rho} \right)^t \frac{\partial U(c_t)}{\partial c_t} \tag{14}$$

$$PK_t = (1-\delta)PK_{t+1} + P_t \frac{\partial F(K_t, L_t)}{\partial K_t} \tag{15}$$

$$P_t = PK_{t+1} \tag{16}$$

donde P_t, PK_t, y PK_{t+1} son los valores de los multiplicadores de Lagrange correspondientes. Estos pueden ser interpretados como el precio del producto, el precio del capital hoy y el precio del capital mañana respectivamente. El problema de maximización anterior está formulado como un problema de NLP. Como ya se ha discutido, MPSGE lo resuelve como un problema MCP. Sean RK_t y W_t la tasa a la que se renta el capital y la tasa de salario en el período t. Denotemos la función de costo unitario

como $C(RK_t, W_t)$ y la función de demanda

como $D(P_t, I)$, en la cual M es el ingreso de consumidor. Así, el MCP puede formularse como sigue.

una función de costo unitario y una función de demanda, respectivamente, en la cual I es el ingreso del consumidor. El modelo, en formato MCP es como sigue:

Condiciones de beneficio cero:

$$P_t \geq PK_{t+1}, I_t \geq 0, I_t[P_t - PK_{t+1}] = 0 \tag{17}$$

$$PK_t \geq RK_t + (1-\delta)PK_{t+1}, K_t \geq 0, K_t[PK_t - RK_t + (1-\delta)PK_{t+1}] = 0 \tag{18}$$

$$C(RK_t, W_t) \geq P_t, Y_t \geq 0, Y_t[C(RK_t, W_t) - P_t] = 0 \tag{19}$$

Condiciones de vaciado de mercados:

$$Y_t \geq D(P_t, M) + I_t, P_t \geq 0, P_t[Y_t - D(P_t, M) + I_t] = 0 \tag{20}$$

$$L_t \geq Y_t \frac{\partial C(RK_t, W_t)}{\partial W_t}, W_t \geq 0, W_t[L_t - Y_t \frac{\partial C(RK_t, W_t)}{\partial W_t}] = 0 \tag{21}$$

$$K_t \geq Y_t \frac{\partial C(RK_t, W_t)}{\partial RK_t}, RK_t \geq 0, RK_t[K_t - Y_t \frac{\partial C(RK_t, W_t)}{\partial RK_t}] = 0 \tag{22}$$

Condición de Balance de Ingresos:

$$M = PK_0 K_0 + \sum_{t=0}^{\infty} W_t L_t, M > 0 \tag{23}$$

Como ya se anotó, el modelista no necesita programar estas condiciones de equilibrio explícitamente. MPSGE las construye automáticamente. Sin embargo necesitamos describir el cambio en el capital a lo largo del tiempo. Esto requiere la modificación de la SAM, presentada en la Tabla1. Adicionaremos un sector de producción adicional para la inversión. Esto, a su vez, habrá de modificar la composición del bloque de bienestar W, porque el nuevo producto podrá ser consumido o invertido, en lugar de ser solamente consumido, como en el caso estático.

En el ejemplo del País de las Maravillas, un estado estacionario requiere que la inversión sea igual a 70. ¿Cómo llegamos a esta cifra?. A fin de cuantificar el valor de la inversión en la senda de crecimiento de estado estacionario, se precisa describir la evolución del capital y del trabajo a lo largo del tiempo. Esto requiere supuestos acerca de la tasa de crecimiento, *g*, la tasa de depreciación del capital, d, y la tasa de interés, *r*. Siendo la fuerza de trabajo inicial L_0, el empleo en el momento *t* es:

$$L_t = L_0(1+g)^t \tag{24}$$

o, en forma equivalente,

$$L_t = (1+g)L_{t-1} \tag{25}$$

La evolución del capital está dada por la ecuación (12). Note que si en el período base una economía se encuentra en una senda de crecimiento de estado estacionario, todas las cantidades (capital, trabajo, producción, consumo) crecen a la misma tasa constante *g* (el caso de no estado estacionario se considera en la siguiente sección). Como tal, la ecuación para el crecimiento del capital puede ser representada de la siguiente manera

$$K_{t+1} = (1+g)K_t \tag{26}$$

En adición supondremos una tasa de interés constante *r*, de forma que todos los precios futuros (incluyendo los del trabajo y el capital) serán, en valor presente:

$$P_{t+1} = \frac{P_t}{1+r} \tag{27}$$

El capital puede ser comprado o rentado. Por consiguiente la implementación de la dinámica involucra dos precios para el capital: el precio de compra, *PK*, y el precio de su arrendamiento, *RK*. Debe enfatizarse que en una SAM estática (Tabla 1), el valor total de las dotaciones de capital, *VK* (observe la intersección *PK* y *CONS* en la Tabla 1 para verificar que *VK* = 100), equivale a sus retornos (*¡y no al acervo!*). La relación entre *VK*, el precio de renta *RK* y el stock de capital es:

$$VK_t = K_t \cdot RK_t \tag{28}$$

Necesitamos ahora considerar las condiciones de primero orden para el capital y la inversión. Rescribámoslas de la siguiente forma:

$$PK_t = (1-\delta)PK_{t+1} + RK_t \tag{29}$$

y

$$PK_{t+1} = P_t \tag{30}$$

La Ecuación (30) puede reordenarse utilizando la ecuación (27) para PK:

$$PK_t = (1+r)P_t \tag{31}$$

Sustituyendo la ecuación (31) para PK_t y la ecuación (30) para PK_{t+1} en (29), tenemos que

$$(1+r)P_t = (1-\delta)P_t + RK_t \tag{32}$$

Como tal, la ecuación del precio de renta del capital es[18]:

$$RK_t = (\delta - g)P_t \tag{33}$$

De las ecuaciones (12) y (26) derivamos la siguiente regla para la inversión de estado estacionario

$$I_t = (\delta + g)K_t \tag{34}$$

o utilizando (28) y (33), la inversión en el período base es igual a

[18] Haciendo P(t)=1, la ecuación para RK tiene perfecto sentido económico porque si el capital y otros activos son sustitutos perfectos, los hogares pueden también recibir la tasa de interés *r* por su préstamo a otros hogares. El capital sin embargo, se deprecia a la tasa δ. Así, *r=RK*-δ*,* o *RK* = δ + *r* que es lo que aparece en la ecuación (33).

$$I_0 = \frac{(\delta + g)VK_0}{\delta + r} \tag{35}$$

La ecuación (35) describe el valor de la inversión que necesitamos introducir en la SAM para una senda de crecimiento de estado estacionario. Asumiremos[19] δ = 0.05, g = 0.02, y r = 0.05. Por consiguiente, la inversión es I = (0.05 + 0.02) . 100/(0.05+0.05) = 70. Supondremos que los bienes se invierten en la misma proporción (35:35). En adición, necesitamos modificar el bloque de bienestar *W* dividiéndolo entre consumo y ahorro. El monto total de bienes producidos es 200, 70 del cual se invierte. Esto deja 130 unidades para el consumo en el periodo base, 65 que constituyen el consumo del bien X y otros 65 que representan el consumo del bien Y. La SAM de referencia luego de estos cambios se presenta en la Tabla 2.

	Sectores Productivos				Consumidores
Mercados	X	Y	W	inv	CONS
PX	100		-65	-35	
PY		100	-65	-35	
PW			200		-200
PL	-40	-60			100
PK	-60	-40			100
sav			-70	70	

Tabla 2. SAM para el modelo dinámico

Una característica importante del problema dinámico es el tratamiento del capital en el último período de modelamiento. No es posible obtener soluciones numéricas para un número infinito de períodos así que se necesita efectuar algunos ajustes para aproximar un modelo de horizonte finito a las elecciones en un horizonte infinito. Deben introducirse procedimientos especiales pare el capital terminal o, en otro caso, todo el capital será consumido en el último período y nada habrá de ser invertido. Hemos seguido el documento de Rutherford, incorporado el nivel del capital posterminal como una variable y adicionado una restricción a la tasa de crecimiento de la inversión en el período terminal:

$$\frac{I_T}{I_{T-1}} = \frac{Y_T}{Y_{T-1}} \tag{36}$$

donde T es el período final. La ventaja del uso de esta restricción es que impone un crecimiento balanceado en el período terminal pero no requiere que el modelo logre el crecimiento de estado estacionario. El significado de la restricción es que la inversión en el período final debe crecer a la misma tasa que el producto[20].

El código del modelo dinámico se presenta en el Apéndice 3 (archivo `move2_1.gms`). A continuación se listan y discuten los cambios que es necesario introducir en el código del modelo estático.

1. Introducción de los conjuntos de tiempo;

2. Declarar las tasas de interés, de crecimiento y de depreciación supuestas;

3. Declarar cuatro o más escalaras: la tasa de renta del capital en el período base, `RK0`, el stock de capital inicial, `K0`, la inversión inicial `I0`, y los retornos del capital iniciales `VK`;

4. Declarar dos parámetros que representen la tasa de crecimiento de las cantidades, `QREF`, y la tasa de crecimiento de los precios `PREF`;

---

[19] Debe tenerse en cuenta que la condición de transversalidad requiere que r > g; en caso contrario, existe la posibilidad de un proceso de préstamos en cadena (para una discusión detallada ver, por ejemplo, Barro y Sala-i-Martin [1995]).

[20] Como todas las cantidades crecen a la misma tasa, es posible utilizar la tasa de crecimiento del producto total, del producto en un cierto sector productivo o del consumo (o simplemente $1 + g$ en el caso de estado estacionario) como variable en el lado derecho de la ecuación (36).

5. Introducir dos nuevos bloques de producción: acumulación de capital, `K`, e inversión, `I`;

6. Cambiar las cantidades de referencia en el bloque `$PROD:W`: 130, 35, 65 en lugar de 200, 100, 100 a fin de representar el ajuste en el nivel de consumo en el período base;

7. Cambiar las cantidades de referencia en el bloque $DEMAND: el consumo en el año base es 130 en lugar 200 y el stock de capital del año base es `K0` en lugar de 100 (*que era una representación del retorno del capital, VK*); ajustar las dotaciones de trabajo y el consumo a las tasas de crecimiento `QREF` y `PREF`;

8. Introducir el capital terminal en los bloques $`AUXILIARY`, $`CONSTRAINT` y $`DEMAND`;

9. Asignar los valores iniciales

Introducción de conjuntos de tiempo.
La introducción de conjuntos de tiempo requiere tres conjuntos `T`, `TFIRST` y `TLAST`; el código GAMS es:

```
SET T /1*10/, TFIRST(T), TLAST(T) ;
TFIRST(T) = YES$(ORD(T) EQ 1);
TLAST(T) = YES$(ORD(T) EQ CARD(T));
```

En la primera linea se declaran los conjuntos. Tenemos 10 períodos. Dada la importancia de los períodos base `TFIRST` y terminal `TLAST`, estos se declaran en conjuntos separados (*subconjuntos de T*). Las dos líneas siguientes asignan los valores para el primer y el último períodos. Esto se hace de una manera que puede parecer un poco complicada para un novicio en GAMS. El propósito de esto es conveniencia pues permite cambiar la dimensión tiempo en una única línea. Por ejemplo, la extensión del modelo a 100 períodos requiere solo de cambiar `T` a `/1*100/`.

Declaración de parámetros supuestos.
Supondremos $g = 0.02$, d = 0.05, y $r = 0.05$. Los siguientes comandos GAMS logran estas declaraciones.

```
SCALAR DELTA      FACTOR DE DESCUENTO          /0.05/
       R          TASA DE INTERES              /0.05/
       G          TASA DE CRECIMIENTO          /0.02/;
```

Declaración de los parámetros iniciales para capital e inversión.
Asignamos directamente el valor del retorno inicial del capital, *VK*, así:

```
SCALAR VK   RETORNO INICIAL DEL CAPITAL        /100.0/;
```

De igual manera, de las ecuaciones (33), (28), y (34) los valores para RK0, K0 e I0 son

```
RK0 = DELTA+R;
K0 = VK/RK0;
I0 = (DELTA + G) * K0;
```

Declaración de tasas de crecimiento para precios y cantidades.
Este paso implementa las ecuaciones (26) y (27) para las tasas de crecimiento en cantidades y precios, que son:

```
QREF(T) = (1 + G) ** (ORD(T) - 1);
PREF(T) = (1/(1 + R)) ** (ORD(T) - 1);
```

Se tiene como exponente `ORD(T)-1` a fin de representar el hecho de que en el año base `QREF` y `PREF` son iguales 1, momento a partir del cual crecen.

Bloques de producción para el capital y la inversión.
El nuevo bloque de producción para K(T) implementa la acumulación de capital según representación en la ecuación (29). El formato MPSGE para este bloque de producción es (omitimos el capital terminal por el momento)

```
$PROD:K(T)
         O:PK(T+1)        Q:((1-DELTA)*K0)
         O:RK(T)          Q:(K0*(DELTA+R))
         I:PK(T)          Q:K0
```

Aquellos que están aprendiendo la representación MPSGE del modelo dinámico de Ramsey suelen confundirse con la cantidad de referencia `RK(T)`. La diferencia con la ecuación (29) es que en el bloque de producción la cantidad de referencia para el precio de renta del capital[21] es K0*(R+DELTA), así como en la ecuación (29) (r+δ) no aparece como coeficiente de RK_t. El modelo dinámico puede no calibrar con esta cantidad de referencia. La razón de esto es el hecho de que *RK* se utiliza en diferentes bloques de producción: *X, Y,* y *K* y necesitamos reflejar la relación entre RK_t y P_t dada por la ecuación (33).

Alternativamente pudimos haber utilizado K0 en referencia del precio para O:RK(T) y asignar el crecimiento del precio de arrendamiento como RK.L(T) = (DELTA+R)*PREF(T);, pero en este caso los campos de entrada de los bloques de producción de X e Y deben también cambiarse según se representa en el programa `move2_2.gms` del Apéndice 3.

El bloque de producción *I* introduce la ecuación (30). Supondremos que la inversión se divide por igual en dos sectores productivos.

```
$PROD:I(T)  S:0
      O:PK(T+1)   Q:I0
      I:PY(T)           Q:(0.5*I0)
      I:PX(T)           Q:(0.5*I0)
```

Note que `S:0` corresponde a la función de producción de Leontief, esto es, la elasticidad de sustitución entre insumos es igual a cero. Si en un bloque de producción no se especifica ninguna elasticidad, MPSGE utiliza cero como valor por defecto. En consecuencia, `S:0` puede ser ignorada en el código de este bloque.

Cambio en el bloque de bienestar.
Necesitamos hacer cambios en la produccióndе *W* porque en la versión dinámica del modelo no todo el producto se consume sino que una parte se dirige a la inversión. De la Tabla 2 obtenemos los nuevos valores para el producto (`Q:130`) y los insumos (`Q:65`).

Cambio en el bloque de demanda.
Por la misma razón, en el bloque DEMAND se necesita modificar la cantidad de referencia correspondiente a PW(T) de 200 a 130. De manera similar, el crecimiento del consumo y el empleo a lo largo del tiempo está representado por `QREF(T)`.

El precio de referencia `PREF(T)` se usa como el valor del descuento sobre el consumo futuro dado por la ecuación (19). La introducción de la preferencia intertemporal de modo individual no es necesaria porque en el modelo de Ramsey, el estado estacionario conduice a un patrón constante de consumo percápita y a la condición según la cual $\rho = r$.

Otro cambio en el bloque `DEMAND` involucra el uso del stock de capital inicial *K*0 como cantidad de referencia para `PK(TFIRST)`, en lugar del valor del capital *VK* como en el caso estático. Esto está determinado por la condición de balance de ingreso del modelo dinámico.

Una modificación más en el bloque `DEMAND` consiste en introducir una elasticidad de sustitución intertemporal `s:1`. Note que hemos asumido el caso Cobb-Douglas en nuestro sencillo modelo dinámico. En

[21] Note que `K0*(R+DELTA)=VK`, de forma que pudimos haber usadi `Q:VK` como cantidad de referencia para `O:RK(T)`.

general, pudimos haber incluido un escalar `SIGMA` como parámetro de elasticidad y hacer `s:SIGMA` (vea el Apéndice 1 para la conversión de funciones de utilidad al formato MPSGE)

Restricción sobre el capital terminal.
Con el fin de introducir una restricción en MPSGE, se necesita utilizar dos tipos de bloques acerca de los cuales no se ha discutido aún. Una restricción se declara en un bloque `$AUXILIARY` y luego se formula en un bloque `$CONSTRAINT`. En nuestro modelo declaramos una restricción *TK*. En adición, se introduce el precio *PKT* para la inclusión de la inversión del último período. Esto representa el mercado del capital posterminal.

El capital posterminal es un producto de los bloques de producción *K* e *I*. La condición `$TLAST(T)` muestra que el capital posterminal se producirá solo en el último período. En los modelos CGE todo aquello que se produce debe ser consumido, de forma que la dotación de capital terminal en el bloque `$DEMAND` se pone en los siguientes términos

```
E:PKT        Q:(-1)       R:TK
```

El valor negativo de la dotación representa el hecho de que el capital terminal habrá de ser consumido. El campo R:TK implica que es necesario satisfacer la siguiente condición:

```
SUM(T$TLAST(T), I(T)/I(T-A) - Y(T)/Y(T-1)) =G= 0;
```

Utilizamos el sumatorio, `SUM`, porque la sintaxis GAMS no permite asignar directamente un elemento de un conjunto a una variable que no tenga el mismo dominio.

Valores Iniciales.
Las asignaciones de efectúan luego de la línea `$SYSINCLUDE` pero antes de la línea `$INCLUDE` del programa. En nuestro programa asignamos valores iniciales para representar los valores presentes (los precios se ajustan con `PREF(T)`, dada por la ecuación (27)) y el crecimiento de las cantidades (que se ajustan por `QREF(T)`, según se representa en la ecuación (26)). La única excepción es `PK.L(T)=(1+R)*PREF(T)`, que simboliza la ecuación (31).

Los valores iniciales para el nivel y el precio del capital terminal se determinan mediante la misma lógica usada para otros precios y cantidades[22]:

```
TK.L = K0 * (1 + G) ** CARD(T);
PKT.L = SUM(TLAST, PK.L(TLAST)/(1+R));
```

El listado de la solución (corra el programa move 2_1.gms para obtenerlo) muestra que todos los marginales son iguales a cero, los niveles de actividad crecen y los precios se reducen a lo largo del tiempo. El consumo total toma el valor de 1145. El precio del capital difiere de todos los otros precios en (1 + r). La Tabla 3 resume los resultados[23]. Luego de replicar la información de referencia, se pueden efectuar experimentos contrafactuales como la introducción de impuestos, cambios en las tasas de crecimiento, elasticidades, etc.

---

[22] Una formulación alternativa es `TK.L = SUM(TLAST, K.L(TLAST));` y `PKT.L = SUM(TLAST, PREF(TLAST));`
[23] Puede ser buena práctica para el aprendiz de MPSGE reproducir los resultados de la Tabla 3.

	Cantidades	Precios	X	Y	W	K	I
1	1.0000	1.0000	100.00	100.00	130.00	1000.00	70.00
2	1.0200	0.9524	102.00	102.00	132.60	1020.00	71.40
3	1.0404	0.9070	104.04	104.04	135.25	1040.40	72.83
4	1.0612	0.8638	106.12	106.12	137.96	1061.21	74.29
5	1.0824	0.8227	108.24	108.24	140.72	1082.43	75.77
6	1.1041	0.7835	110.41	110.41	143.53	1104.08	77.29
7	1.1262	0.7462	112.62	112.62	146.40	1126.16	78.83
8	1.1487	0.7107	114.87	114.87	149.33	1149.69	80.41
9	1.1717	0.6768	117.17	117.17	152.32	1172.66	82.02
10	1.1951	0.6446	119.51	119.51	155.36	1195.09	83.66

Tabla 3. Precios y cantidades de estado estacionario

4.2. Calibración de Modelos Dinámicos con información de referencia no consistente con estado estacionario

La información para el período base puede no ser consistente con una senda de crecimiento de estado estacionario. La Tabla 4 da un ejemplo de una situación en la cual la inversión es "muy grande", esto es, de la ecuación (35) el valor de la inversión es 80 > (0.05 + 0.02) . 100/(0.05 + 0.05) = 70.

	Sectores Productivos				Consumidores
Mercados	X	Y	W	inv	CONS
PX	100		-60	-40	
PY		100	-60	-40	
PW			200		-200
PL	-40	-60			100
PK	-60	-40			100
sav			-80	80	

Tabla 4. SAM para un modelo dinámico fuera del estado estacionario

Una posible solución para la calibración del modelo en el caso de información no consistente con el estado estacionario, es asumir que *g* es la tasa de crecimiento del trabajo y que la tasa de crecimiento del capital es inicialmente diferente, aunque converge a *g* a lo largo del tiempo[24]. Usaremos el método de calibración basado en el algoritmo de la secante. Este encuentra la solución de una ecuación $f(x) = 0$ a partir de:

1. Estimación inicial de $x(n)$, donde n es el número de la iteración.
2. Calculo de $x(n + 1)$ como[25]:

[24] Otra forma de calibrar con datos estáticos implica la información dada sobre inversión, tasa de interés y retrono del capital para un año base. La tasa de interés de largo plazo en el estado estacionario está determinada por la tasa de crecimiento del empleo asumida y la tasa de descuento calibrada.

[25] La ecuación (37) puede derivarse de la ecuación de la tangente: $\frac{y - y_n}{x - x_n} = f'(x_n)$. Haga $y = 0$ y $y_n = f(x_n)$, por lo que $x = x_n - \frac{f(x_n)}{f'(x_n)}$. Usando $f'(x_n) = \frac{f(x_n) - f(x_{n-1})}{x_n - x_{n-1}}$ obtenemos la ecuación (37).

$$x_{n+1} = x_n - \frac{f(x_n)}{f(x_n) - f(x_{n-1})}(x_n - x_{n-1}) \tag{37}$$

Para reiterar nuestro problema dinámico conocemos los valores iniciales en la SAM, que no es consistente con la senda de crecimiento de estado estacionario. Adicionalmente conocemos (las hemos supuesto) la tasa de depreciación y la de interés de largo plazo, si bien el precio de renta permanece desconocido en el período base. A fin de obtener este valor haremos iteraciones basadas en el algoritmo líneas arriba. La calibración supone el siguiente proceso de iteración: estimar la tasa de renta de capital inicial *RK*o, calcular un modelo de la senda de transición, observar la inversión del equilibrio actual, *I*o, y ajustar *RK*o.

El código del programa para el modelo dinámico del caso de no estado estacionario se encuentra en el Apéndice 3 (archivo `move3_1.gms`). Inicialmente, hacemos una suposición acerca del precio de renta del capital, RKG, en el para el primer período. En el caso de estado estacionario el precio de renta (que también es igual a la renta del capital) es igual a $\delta + r$. Haremos la suposición basados en ese dato. Note que si nuestra suposición se encuentra "muy lejos", el algoritmo no funcionará. En forma paralela, si la inversión está "demasiado lejos" del valor de estado estacionario, el algoritmo tampoco funcionará.

El programa dinámico de no estado estacionario tiene la misma estructura que antes excepto por el hecho de que hemos usado una formulación alternativa similar a la del programa descrito en el Apéndice 3 (archivo `move2_2.gms`). Aplicamos la ecuación (37) para hallar $f(x)$ que, en nuestro caso, es la diferencia entre el valor de la inversión calculada en el programa y la que se deriva de la SAM. De forma similar, $x(n)$ es la suposición sobre el precio de renta en el primer período. Durante las primeras iteraciones simplemente sumamos 0.1 a nuestro valor supuesto para *RK*o a fin de producir una base para futuras iteraciones. Durante todas las iteraciones restantes, reproducimos la ecuación (37) así:

```
RKG = RKG + ERROR(ITER)/(ERROR(ITER-1)-ERROR(ITER))
                                * (RK0VAL(ITER)-RK0VAL(ITER-1));
```

donde `ERROR` representa[26] $f(x)$ y `RK0VAL` representa $x(n)$.

La porción final del programa incluye graficar los resultados para la inversión, el consumo y las tasas de crecimiento y de interés. Usamos GNUPLOT para producir los gráficos. Puede observarse en estos (corra `move3_1.gms` para obtener los gráficos) que los valores convergen eventualmente a una senda de crecimiento de estado estacionario. La velocidad de convergencia está determinada por los parámetros de la producción y las preferencias. (Barro y Sala-i-Martin [1995]).

4.2.Dinámica en el GTAP

Ajustes similares a los descritos arriba pueden ser aplicados a la versión estática del paquete GTAPinGAMS (Rutherford, [1998b]). El Apéndice 4 presenta el código del modelo dinámico.

5. Conclusión

¡Aprenda MPSGE!!!

Referencias

[1] Arrow, K.J., and G. Debreu (1954). "Existence of an Equilibrium for a Competitive Economy." *Econometrica,* 22, 265-90.

[26] Note que para $f(x)$ utilizamos una ecuación para el consumo en lugar de una para inversión a fin de separar explicitamente los valores dados de los calculados. Al usar `ERROR(ITER) = I.L("1")*I0 - I0;` en lugar de `ERROR(ITER) = C.L("1") - (200-I0);` obtenemos los mismos resultados.

[2] Barro, R.J., and X. Sala-i-Martin (1995). *Economic Growth,* McGraw-Hill.

[3] Brooke, A., D. Kendrick, and A, Meeraus (1992). *GAMS: A User's Guide, Release 2.25,* Scientific Press.

[4] Ginsburgh V., and M. Keyzer (1997). *The Structure of Applied General Equilibrium Models,* The MIT Press.

[5] Lau M.i., A. Pahlke, and T.F. Rutherford (1997). "Modeling Economic Adjustment: A Primer in Dynamic General Equilibrium Analysis", University of Colorado. Working Paper, available at: http://nash.colorado.edu/tomruth/primer/paper.htm

[6] Markusen, J.R., and T.F. Rutherford (1995). The Markusen examples. Notes for GAMS workshop, 1995. (available at http://nash.colorado.edu/tomruth/6433/markusen.htm)

[7] Mathiesen, L. (1985). "Computation of Economic Equilibrium by a Sequence of Linear Complementarity Problems", *Mathematica Programming Study 23,* North-Holland, pp. 144-162.

[8] Rutherford, T.F. (1998). "Overlapping Generations with Pure Exchange: An MPSGE Formulation", University of Colorado, mimeo, available at: http://debreu.colorado.edu/tomruth/olg/exchange/indeml

[9] Rutherford, T.F. (1998b). "GTAPinGAMS: The Dataset and Static Model", University of Colorado, available at: http://debreu.colorado.edu/tomruth/gtampingams/htmlgtapgams.html

[10] Rutherford, T.F. (1999). "Applied General Equilibrium Modeling with MPSGE as a GAMS Subsystem: An overview of the Modeling Framework and Syntax", *Computational Economics,* V.14, Nos. 1-2.

[11] Rutherford, T.F., and S.V. Paltsev (1999). From an Input-Output Table to a General Equilibrium Model: Assessing the Excess Burden of Indirect Taxes in Russia. University of Colorado. Working Paper. (available at http:/nash.colorado.edu/gams-x/ruswebfiles/gams-cgi-ststst.html)

[12] Varian, H.R. (1992). *Microeconomic Analysis,* Third Edition, Norton & Company.

Apéndice 1. Conversión de funciones de producción y utilidad al formato MPSGE

MPSGE representa funciones de producción y de preferencias como funciones CES anidadas.

Producción:
Una tecnología de producción se representa idénticamente tanto por una función de producción como por una función de costos. Una forma general para la función CES anidada es $X(K,L)=\left(\alpha K^{\rho}+(1-\alpha)L^{\rho}\right)^{\frac{1}{\rho}}$ donde ρ es el parámetro de sustitución ($-\infty < \rho < 1$) cuya relación con la elasticidad de sustitución σ es $\sigma = \frac{1}{1-\rho}$. Cuando $\rho = -\infty$ (o cuando $\sigma = 0$) la función de producción CES se torna en una Leontief (o de proporciones fijas) $X_1(K,L)=\min(aK_1,(1-a)L_1)$. Cuando $\rho = 1$ (cuando $\sigma = \infty$), los insumos son sustitutos perfectos en la producción $X_2 = aK_2 + (1-a)L_2$. Cuando $\rho = 0$ (cuando $\sigma = 1$), la función de producción CES se convierte en un caso Cobb-Douglas $X_3(K,L)=K_3^{a}L_3^{1-a}$.

Considérese la representación MPSGE de un ejemplo particular de la función de producción Cobb-Douglas: $X(K,L)=K^{0.6}L^{0.4}$. Los siguientes bloques de producción representan la misma tecnología:

a)

```
$PROD:X     s:1
      O:PX  Q:1
      I:PK  Q:0.6
      I:PL  Q:0.4
```

b)

```
$PROD:X     s:1
      O:PX  Q:100
      I:PK  Q:60
      I:PL  Q:40
```

c)

```
$PROD:X     s:1
      O:PX  Q:100
      I:PK  Q:60  P:2
      I:PL  Q:40  P:2
```

d)

```
$PROD:X     s:1
      O:PX  Q:1   P:2
      I:PK  Q:60
      I:PL  Q:40
```

e)

```
$PROD:X     s:1
      O:PX  Q:100 P:2
      I:PK  Q:60  P:2
```

```
      I:PL  Q:40  P:2
```

Recuerde que los precios y las cantidades de referencia solo se utilizan el la calibración y no se transfieren al *solver* como valores iniciales! Otro asunto de importancia es que solo los precios relativos tienen relevancia. Como tal el uso de `P:1` o `P:2` para los dos insumos da la misma tasa marginal de sustitución. El precio del producto no es de importancia en este ejemplo porque tenemos un solo bien producido. El caso de dos productos, los precios relativos si interesan pues definen la tasa marginal de transformación.

¿Qué tal el siguiente bloque de producción?

```
$PROD:X     s:1
      O:PX  Q:1
      I:PK  Q:1   P:0.6
      I:PL  Q:1   P:0.4
```

Representa la misma tecnología (?) de los bloques de producción MPSGE anteriores.

Ahora considérese el bloque de producción:

```
$PROD:X     s:1
      O:PX  Q:1
      I:PK  Q:0.6 P:2
      I:PL  Q:0.4
```

Este bloque está descrito por la función $X(K,L)=K^{0.75}L^{0.25}$, por que los valores de distribución son: $K=\frac{1.2}{1.2+0.4}=0.75$ y $L=\frac{0.4}{1.2+0.4}=0.25$

La función de producción Leontief puede ser representada por cualquiera de los dos bloques a continuación:

```
$PROD:X     s:0
      O:PX  Q:100
      I:PK  Q:60
      I:PL  Q:40
```

o bien:

```
$PROD:X
      O:PX  Q:100
      I:PK  Q:60
      I:PL  Q:40
```

Dado que el valor por defecto de la elasticidad en el nivel superior es cero.

Preferencias:

La forma general de una función de utilidad CES anidada es $U(X,Y)=\left(\alpha X^{\rho}+(1-\alpha)Y^{\rho}\right)^{\frac{1}{\rho}}$, donde ρ es un parámetro de elasticidad. La elasticidad de sustitución σ se relaciona con el parámetro de elasticidad mediante $\sigma=\frac{1}{1-\rho}$. El bloque `$DEMAND` representa la utilidad derivada del consumo de mercancías y dotaciones. Si, por ejemplo todas las cantidades demandadas así como las son iguales a 100, el bloque de demanda en MPSGE tomará la siguiente forma:

```
$DEMAND:CONS     s:(1/1-RHO)
      D:PX       Q:100
      D:PY       Q:100
```

```
        E:PL        Q:100
        E:PK        Q:100
```

Existe una función de utilidad ampliamente utilizada en el análisis dinámico. Se trata de la función de utilidad de elasticidad de sustitución intertemporal constante (CIES) que tiene la siguiente forma: $U(C_t)=\frac{C_t^{1-\theta}-1}{1-\theta}$, donde $\sigma=\frac{1}{\theta}$ es la elasticidad de sustitución.

Cuando $\theta\rightarrow 0$, $U(C_t)$ se aproxima al caso lineal $U(C_t)$ = C_t - 1 (que representa las mismas preferencias que $U(C_t) = C_t$). Cuando $\theta\rightarrow 1$, $U(C_t) = ln(C_t)$.

A fin de entender la relación entre las formas CES y CIES, empezaremos con una función CES: $U(C_t)=(\alpha C_t^{\rho})^{\frac{1}{\rho}}$. Haciendo $\rho=1-\theta$ dado que $\sigma=\frac{1}{\theta}$ y que $\sigma=\frac{1}{1-\rho}$ se llega a $U(C_t)=(\alpha C_t^{1-\rho})^{\frac{1}{1-\rho}}$.La maximización de $(\alpha C_t^{1-\rho})^{\frac{1}{1-\rho}}$ y de $\frac{\alpha C^{1-\theta}}{1-\theta}$ produce el mismo valor óptimo para C_t (puesto que max $X^{\frac{1}{b}}\Leftrightarrow$ max $X^b\Leftrightarrow$ max $\frac{X}{b}$). Para la maximización de la utilidad sobre un horizonte infinito, sumaremos las utilidades instantáneas y ajustaremos con un factor de descuento, cuestión que conduce a la forma tradicional de la función de utilidad:

$$U=\sum_{t=0}^{\infty}\left(\frac{1}{1+\rho}\right)\frac{C_t^{1-\theta}}{1-\theta}$$

Si un consumidor tiene una dotación de trabajo y capital y una elasticidad de sustitución intertemporal `SIGMA` entre los consumos de los distintos períodos, definida por `SIGMA=1/THETA`, a fin de reflejar la relación entre σ y θ, el bloque `$DEMAND` para la función de utilidad CIES tendrá la siguiente forma:

```
$DEMAND:CONS s:SIGMA
        D:PC(T)             Q:QREF(T)    P:PREF(T)
        E:PL(T)             Q:QREF(T)
        E:PK(TFIRST)        Q:K0
```

Note que si el consumidor obtiene utilidad de dos bienes (por ejemplo en una forma Cobb-Douglas), se puede mantener el mismo bloque `$DEMAND`, adicionando un bloque de producción:

```
$PROD:C(T) s:1
        O:PC(T)             Q:C0
        I:PX(T)             Q:X0
        I:PY(T)             Q:Y0
```

Anidamientos Multinivel

Considere el caso para el cual T=1,2. Entonces, en los siguientes bloques de producción:

a)

```
$PROD:C      s:1
        O:PC                Q:C0
        I:PX(T)             Q:X0
        I:PY(T)             Q:Y0
```

La elasticidad entre `PX1, PX2, PY1, PY2` es 1.

b)

```
$PROD:C      s:1        a:0.5
      O:PC              Q:C0
      I:PX(T)           Q:X0        a:
      I:PY(T)           Q:Y0
```

La elasticidad entre `PX1` y `PX2` es 0.5, y la elasticidad entre `PY1`, `PY2` y `PX1:PX2` es 1.

c)

```
$PROD:C      s:1        a:0.5       b:1.5
      O:PC              Q:C0
      I:PX(T)           Q:X0        a:
      I:PY(T)           Q:Y0        b:
```

La elasticidad entre `PX1` y `PX2` es 0.5, la elasticidad entre `PY1` y `PY2` es 1.5, y la elasticidad entre `PX1:PX2` y `PY1:PY2` es 1.

d)

```
$PROD:C      s:1        a:0.5
      O:PC              Q:C0
      I:PX(T)           Q:X0        a:
      I:PY(T)           Q:Y0        a:
```

La elasticidad el nivel superior no importa y la elasticidad (anidada) entre `PX1:PX2:PY1:PY2` es 0.5

e)

```
$PROD:C      s:1        T.tl:0.5
      O:PC              Q:C0
      I:PX(T)           Q:X0        T.tl:
      I:PY(T)           Q:Y0        T.tl:
```

La elasticidad entre `PX1` y `PY1` es 0.5, la elasticidad entre `PX2` y `PY2` es 0.5, y la elasticidad entre `PX1:PY1` y `PX2:PY2` es 1.

Apéndice 2. MPSGE y la formulación algebraica

EL apéndice 2 contiene las versiones MPSGE y algebraica (MCP) para el siguiente problema (Shoven and Whalley, *Applied General-Equilibrium Models of Taxation and Internacional Trade: An Introduction and Survey*, Journal of Economic Literature, Vol 22, Nº 3 (Sep., 1984), pp. 1007-1051). Se proporciona a efectos de ilustración: permite comparar el grado de dificultad de las dos versiones. Si le interesa la versión MCP del modelo de Ramsey, diríjase al documento de Rutherford [1997]. Tengo algunas notas preliminares acerca de la conversión de MPSGE a MCP. Hablemos luego para ver si las he terminado ya.

En este modelo hay dos bienes finales (M,N), dos factores de producción (K,L) y dos consumidores (R,P). Los productores son competitivos, minimizan costes unitarios tomando los precios del merado como dados. Las funciones de producción son del tipo de elasticidad de sustitución constante, (CES):

$$Q_i = \phi_i \left[\delta_i L_i^{\rho_i} + (1-\delta_i) K_i^{\rho_i} \right]^{1/\rho_i} \qquad i \,\varepsilon\, M,N$$

donde $\rho_i = (\sigma_i - 1)/\sigma_i$, y σ_i es una elasticidad de sustitución entre trabajo y capital. Los parámetros de la función de producción son:

Sector	ϕ_i	δ_i	σ_i
Manufacturero (M)	1.6	0.6	2.0
No Manufacturero (N)	2.0	0.7	0.5

Los consumidores cuentan con una dotación de factores primarios de los cuales obtienen su ingreso. Este es utilizado en la compra de bienes con el fin de maximizar la utilidad. Las funciones de utilidad CES tienen la siguiente forma:

$$U_c = \left[\sum_i \alpha_{ic}^{1/\sigma_c} X_{ic}^{\rho_c} \right]^{1/\rho_c} \qquad c \,\varepsilon\, R,P$$

donde $\rho_c = (\sigma_c - 1)/\sigma_c$, y σ_c es una elasticidad de sustitución entre los bienes M y N. Las dotaciones de bienes y los parámetros de la función de utilidad son:

Hogares	K	L	α_M	α_N	σ_c
Rico	25	0	0.5	0.5	1.5
Pobre	0	60	0.3	0.7	0.75

Luego de calcular el equilibrio del *benchmark* se efectúa un experimento contrafactual en el cual se impone un impuesto *ad-valorem* del 50% sobre cada unidad de servicios de capital aplicados en el sector manufacturero. Los ingresos fiscales se retornan *lump-sum* a los hogares: 40% a los ricos y 60% a los pobres.

```
$TITLE MPSGE y la representacion algebraica

SET                      i       Tipo de Bien     /M      Manufacturado
                                                   N      No Manufacturado /

                         c       Tipo de Hogar    /Rico   Hogar Rico
                                                   Pobre  Hogar Pobre/

                    rico(c)                       /Rico/

                   pobre(c)                       /Pobre/;
```

```
PARAMETERS
         sigma           /m 2, n 0.5/,
         phi             /m 1.5, n 2/,
         delta           /m 0.6, n 0.7/,
         sig             /rico 1.5, pobre 0.75/,
         capital         /rico 25, pobre 0/,
         labor           /rico 0, pobre 60/;

TABLE    alpha(c,i)

          M      N
RICO    0.5    0.5
POBRE   0.3    0.7 ;


PARAMETER
         taxrate(i)

$ONTEXT

$MODEL:WORK

$SECTORS:
         Y(i)            !Produccion

$COMMODITIES:
         P(i)            !Precio de la Produccion
         W               !Precio del Trabajo
         R               !Precio del Capital

$CONSUMERS:
         cons(c)         !Agente Representativo

$PROD:y(i)       s:sigma(i)
         o:p(i)          q:phi(i)
         i:w q:1         p:delta(i)
         i:r q:1         p:(1-delta(i))
+        a:cons("RICO")         t:(0.4*taxrate(i))
+        a:cons("POBRE")         t:(0.6*taxrate(i))


$demand:cons(c) s:sig(c)
         d:p(i)          q:1 p:(alpha(c,i)**(1/sig(c)))
         e:r             q:capital(c)
         e:w             q:labor(c)

$OFFTEXT

$SYSINCLUDE mpsgeset work

taxrate(i) = 0;

work.iterlim = 5000;

$include work.gen

SOLVE work USING MCP;

taxrate("M") = 0.5;

$include work.gen
```

```
SOLVE work USING mcp;

*-----------------------------------------------------------------------------
* Version MCP
*-----------------------------------------------------------------------------

ALIAS(i,j);

taxrate("M") = 0;

PARAMETER pk0(i);

pk0(i) = 1+taxrate(i);

PARAMETER thetak(i), thetal(i);

         thetak(i) = 1-delta(i);
         thetal(i) = delta(i);

PARAMETER abeta(c,i);

         abeta(c,i) = alpha(c,i)**(1/sig(c));

EQUATIONS
         prf_y(i)        Condicion de Beneficio Cero
         mkt_y(i)        Vaciado del Mercado del Producto
         mkt_k           Vaciado del Mercado del Capital
         mkt_l           Vaciado del Mercado del Trabajo
         i_cons(c)       Condicion de Equilibrio de Ingresos;

* Condicion de Beneficio Cero

prf_y(i)..
         (thetak(i)*(r*(1+taxrate(i))/(pk0(i)*(1-delta(i))))**(1-sigma(i))
         + thetal(i)*(w/delta(i))**(1-sigma(i)))**(1/(1-sigma(i)))
         =e= p(i) * phi(i);

* Vaciado de Mercados

mkt_y(i)..
         y(i) * phi(i) =e=
         sum(c, 1 *
         (((sum(j, abeta(c,j) * (p(j)/abeta(c,j))
         ** (1-sig(c))))**(1/(1-sig(c)))) * abeta(c,i) / p(i)) ** sig(c)) *
         cons(c) / ( ((sum(j, abeta(c,j) * (p(j)/abeta(c,j))
         ** (1-sig(c))))**(1/(1-sig(c))))) );

mkt_k..
         sum(c, capital(c)) =e=
         sum(i,
         ((thetal(i)*(w/delta(i))**(1-sigma(i))
         + thetak(i)*(r*(1+taxrate(i))/(pk0(i)*(1-delta(i))))**(1-sigma(i)))
         **(1/(1-sigma(i))) / (r*(1+taxrate(i))/(pk0(i)*(1-delta(i)))))
         **(sigma(i)) * y(i));

mkt_l..
         sum(c, labor(c)) =e=
         sum(i,
         ((thetal(i)*(w/delta(i))**(1-sigma(i))
         + thetak(i)*(r*(1+taxrate(i))/(pk0(i)*(1-delta(i))))**(1-sigma(i)))
         **(1/(1-sigma(i))) * delta(i)/w)**(sigma(i)) * y(i));

i_cons(c)..
```

```
         cons(c) =e= r*capital(c) + w*labor(c)
         + (sum(i, 0.6 * taxrate(i) * y(i) * r *
         ((thetal(i)*(w/delta(i))**(1-sigma(i))
         + thetak(i)*(r*(1+taxrate(i))/(pk0(i)*(1-delta(i))))**(1-sigma(i)))
         **(1/(1-sigma(i)))
         / (r*(1+taxrate(i))/(pk0(i)*(1-delta(i)))))**(sigma(i))))$pobre(c)
         + (sum(i, 0.4 * taxrate(i) * y(i) * r *
         ((thetal(i)*(w/delta(i))**(1-sigma(i))
         + thetak(i)*(r*(1+taxrate(i))/(pk0(i)*(1-delta(i))))**(1-sigma(i)))
         **(1/(1-sigma(i)))
         / (r*(1+taxrate(i))/(pk0(i)*(1-delta(i)))))**(sigma(i))))$rico(c);

MODEL obana /prf_y.y, mkt_y.p, mkt_k.r, mkt_l.w, i_cons.cons/;

cons.fx("rico") = 34.3368;
cons.fx("pobre") = 60.0000;
obana.iterlim = 5000;

SOLVE obana USING mcp;

* Intruduccion del Impuesto

taxrate("M") = 0.5;
cons.fx("rico") = 29.0935;
cons.fx("pobre") = 61.3484;

SOLVE obana using mcp;
```

Apéndice 3. Listados de los Programas

El Apéndice 3 contiene programas GAMS que pueden ser copiados y ejecutados. Se localizan aquí los siguientes archivos de programa:

move1_1.gms	Modelo Estático	El ejemplo del País de las Maravillas
move1_2.gms		Sin el bloque de bienestar
move1_2.gms		Cantidades iniciales
move1_4.gms		Bloque de reportes
move1_5.gms		Sintaxis vectorial
move2_1.gms	Modelo Dinámico	En estado estacionario en el año base
move2_2.gms		Formulación alternativa para el bloque de K
move2_3.gms		Fuera del estado estacionario en el año base

```
$TITLE: move1_1.gms - Un Modelo Estatico Simple (El ejemplo del Pais de las
Maravillas)

$ONTEXT

$MODEL:MOVE1_1

$SECTORS:
        X        ! Nivel de actividad del sector X
        Y        ! Nivel de actividad del sector Y
        W        ! Nivel de actividad del sector W (indice de bienestar)

$COMMODITIES:
        PX       ! Indice de precios del bien X
        PY       ! Indice de precios del bien Y
        PL       ! Indice de precios del factor primario L
        PK       ! Indice de precios del factor primario K
        PW       ! Indice de precios para el bienestar (funcion de gasto)

$CONSUMERS:
        CONS     ! Nivel de Ingreso del consumidor CONS

$PROD:X s:1
        O:PX    Q:100
        I:PL    Q:40
        I:PK    Q:60

$PROD:Y s:1
        O:PY    Q:100
        I:PL    Q:60
        I:PK    Q:40

$PROD:W s:1
        O:PW    Q:200
        I:PX    Q:100
        I:PY    Q:100

$DEMAND:CONS
        D:PW    Q:200
        E:PL    Q:100
        E:PK    Q:100

$OFFTEXT

$SYSINCLUDE mpsgeset MOVE1_1

MOVE1_1.ITERLIM = 0;

$INCLUDE MOVE1_1.GEN

SOLVE MOVE1_1 USING MCP;
```

```
$TITLE: move1_2.gms - Un modelo estatico simple (sin el bloque de bienestar)

$ONTEXT

$MODEL:MOVE1_2

$SECTORS:
         X       ! Nivel de Actividad del Sector X
         Y       ! Nivel de Actividad del Sector Y

$COMMODITIES:
         PX      ! Indice de Precios - bien X
         PY      ! Indice de Precios - bien Y
         PL      ! Indice de Precios - factor primario L
         PK      ! Indice de Precios - factor primario K

$CONSUMERS:
         CONS    ! Nivel de Ingreso del consumidor CONS

$PROD:X s:1
         O:PX    Q:100
         I:PL    Q:40
         I:PK    Q:60

$PROD:Y s:1
         O:PY    Q:100
         I:PL    Q:60
         I:PK    Q:40

$DEMAND:CONS s:1
         D:PX    Q:100
         D:PY    Q:100
         E:PL    Q:100
         E:PK    Q:100

$OFFTEXT

$SYSINCLUDE mpsgeset MOVE1_2

MOVE1_2.ITERLIM = 0;

$INCLUDE MOVE1_2.GEN

SOLVE MOVE1_2 USING MCP;
```

```
$TITLE: move1_3.gms - Modelo Estatico Simple: Formulacion para las cantidades
iniciales

$ONTEXT

$MODEL:MOVE1_3

$SECTORS:
        X        ! Nivel de actividad del sector X
        Y        ! Nivel de actividad del sector Y
        W        ! Nivel de actividad del sector W (indice de bienestar)

$COMMODITIES:
        PX       ! Indice de precios del bien X
        PY       ! Indice de precios del bien Y
        PL       ! Indice de precios del factor primario L
        PK       ! Indice de precios del factor primario K
        PW       ! Indice de precios para el bienestar (funcion gasto)

$CONSUMERS:
        CONS     ! Nivel de Ingreso del consumidor CONS

$PROD:X s:1
        O:PX    Q:1
        I:PL    Q:0.4
        I:PK    Q:0.6

$PROD:Y s:1
        O:PY    Q:1
        I:PL    Q:0.6
        I:PK    Q:0.4

$PROD:W s:1
        O:PW    Q:1
        I:PX    Q:0.5
        I:PY    Q:0.5

$DEMAND:CONS
        D:PW    Q:200
        E:PL    Q:100
        E:PK    Q:100

$OFFTEXT

$SYSINCLUDE mpsgeset MOVE1_3

X.L=100;
Y.L=100;
W.L=200;

MOVE1_3.ITERLIM = 0;

$INCLUDE MOVE1_3.GEN

SOLVE MOVE1_3 USING MCP;
```

```
$TITLE: move1_4.gms - Modelo Estatico Simple (con bloque de reportes)

$ONTEXT

$MODEL:MOVE1_4

$SECTORS:
        X        ! Nivel de actividad del sector X
        Y        ! Nivel de actividad del sector Y
        W        ! Nivel de actividad del sector W (indice de bienestar)

$COMMODITIES:
        PX       ! Indice de precios del bien X
        PY       ! Indice de precios del bien Y
        PL       ! Indice de precios del factor primario L
        PK       ! Indice de precios del factor ptimario K
        PW       ! Indice de precios para el bienestar (funcion gasto)

$CONSUMERS:
        CONS     ! Nivel de Ingreso del consumidor CONS

$PROD:X s:1
        O:PX    Q:100
        I:PL    Q:40
        I:PK    Q:60

$PROD:Y s:1
        O:PY    Q:100
        I:PL    Q:60
        I:PK    Q:40

$PROD:W s:1
        O:PW    Q:200
        I:PX    Q:100
        I:PY    Q:100

$DEMAND:CONS
        D:PW    Q:200
        E:PL    Q:100
        E:PK    Q:100

$REPORT:
        V:X_OUT    O:PX    PROD:X
        V:K_X      I:PK    PROD:X
        V:L_X      I:PL    PROD:X
        V:Y_OUT    O:PY    PROD:Y
        V:K_Y      I:PK    PROD:Y
        V:L_Y      I:PL    PROD:Y
        V:W_OUT    O:PW    PROD:W
        V:X_W      I:PX    PROD:W
        V:Y_W      I:PY    PROD:W
        V:C        D:PW    DEMAND:CONS

$OFFTEXT

$SYSINCLUDE mpsgeset MOVE1_4

MOVE1_4.ITERLIM = 0;
$INCLUDE MOVE1_4.GEN
SOLVE MOVE1_4 USING MCP;
```

```
PARAMETER reporte;

reporte("px","X") = x_out.l;
reporte("py","y") = y_out.l;
reporte("pw","w") = w_out.l;
reporte("pl","X") = -l_x.l;
reporte("pk","X") = -k_x.l;
reporte("pl","y") = -l_y.l;
reporte("pk","y") = -k_y.l;
reporte("px","w") = -x_w.l;
reporte("py","w") = -y_w.l;
reporte("pl","cons") = l_x.l+l_y.l;
reporte("pk","cons") = k_x.l+k_y.l;
reporte("pw","cons") = -w_out.l;

DISPLAY reporte;
```

```
$TITLE: move1_5.gms - Un Modelo Estatico Simple (sintaxis vectorial)


SET              i       Sectores        /X,Y/
                 f       Factores        /K,L/;


TABLE sam(*,*) Informacion de Benchmark

         X    Y     W  CONS
X      100       -100
Y           100  -100
W                 200  -200
L      -40  -60         100
K      -60  -40         100 ;


PARAMETER        Y0(I)           Benchmark sectoral output,
                 FD0(F,I)        Benchmark factor demands,
                 C0(I)           Benchmark consumption demand,
                 E0(F)           Factor endowments,
                 W0              Benchmark total consumption;


* Se toman datos del formato original y se disponen en vectores adecuados

       Y0(I) =  SAM(I,I);
    FD0(F,I) = -SAM(F,I);
       C0(I) = -SAM(I,"W");
         W0  =  SUM(I, C0(I));
       E0(F) =  SAM(F,"CONS");

$ONTEXT

$MODEL:MOVE1_5

$SECTORS:
         OUT(i)          ! Production Activity Level
         W               ! Welfare Index

$COMMODITIES:
         P(i)            ! Price index for commodities
         PF(f)           ! Price index for factors
         PW              ! Utility Price Index

$CONSUMERS:
         CONS            ! Income level for consumer CONS

$PROD:OUT(i)
         O:P(i)           Q:Y0(i)
         I:PF(f)          Q:FD0(f,i)

$PROD:W s:1
         O:PW             Q:W0
         I:P(i)           Q:C0(i)

$DEMAND:CONS
         D:PW             Q:W0
         E:PF(f)          Q:E0(f)
```

```
$REPORT:
        V:OUT_o(i)       O:P(i)         PROD:OUT(i)
        V:INP(f,i)       I:PF(f)        PROD:OUT(i)
        V:OUT_w          O:PW           PROD:W
        V:INP_w(i)       I:P(i)         PROD:W
        V:C              D:PW           DEMAND:CONS

$OFFTEXT

$SYSINCLUDE mpsgeset MOVE1_5

MOVE1_5.ITERLIM = 0;

$INCLUDE MOVE1_5.GEN

SOLVE MOVE1_5 USING MCP;

PARAMETER report;
         report(i,i)        =  out_o.l(i);
         report("w","w")    =  out_w.l;
         report(i,"w")      = -inp_w.l(i);
         report(f,i)        =  inp.l(f,i);
         report(f,"cons")   =  sum(i, inp.l(f,i));
         report("w","cons") = -out_w.l;
DISPLAY report;
```

```
$TITLE: move2_1.gms. Modelo de Crecimiento de Ramsey Simple

SET      T       /1*10/,
         TFIRST(T),
         TLAST(T);

TFIRST(T) = YES$(ORD(T) EQ 1);
TLAST(T)  = YES$(ORD(T) EQ CARD(T));

SCALAR   DELTA   FACTOR DE DESCUENTO                /0.05/
         R       TASA DE INTERES                    /0.05/
         G       TASA DE CRECIMIENTO                /0.02/
         VK      INGRESOS INICIALES DEL CAPITAL     /100.0/
         K0      STOCK INICIAL DE CAPITAL
         RK0     RETORNO INICIAL DEL CAPITAL
         I0      INVERSION INICIAL;

PARAMETER        QREF(T) CANTIDADES
                 PREF(T) PRECIOS;

RK0 = DELTA+R;
K0  = VK/RK0;
I0  = (DELTA + G) * K0;
QREF(T) = (1 + G) ** (ORD(T) - 1);
PREF(T) = (1/(1+R)) ** (ORD(T) - 1);

$ONTEXT

$MODEL:MOVE2_1

$SECTORS:
         X(T)    ! Nivel de actividad del sector X
         Y(T)    ! Nivel de actividad del sector Y
         W(T)    ! Nivel de actividad del sector W (indice de bienestar)
         I(T)    ! Sector Inversion
         K(T)    ! Acumulacion de Capital

$COMMODITIES:
         PX(T)   ! Indice de precios para el bien X
         PY(T)   ! Indice de precios del bien Y
         PL(T)   ! Indice de precios del factor primario L
         PK(T)   ! Indice de precios del factor primario K
         PW(T)   ! indice de precios del bienestar (funcion gasto)
         RK(T)   ! Precio de renta del capital
         PKT     ! Restriccion sobre el capital posterminal

$CONSUMERS:
         CONS    ! Nivel de ingreso del consumidor CONS

$AUXILIARY:
         TK      ! Stock de capital terminal

$PROD:X(T) s:1
         O:PX(T)        Q:100
         I:PL(T)        Q:40
         I:RK(T)        Q:60

$PROD:Y(T) s:1
         O:PY(T)        Q:100
         I:PL(T)        Q:60
```

```
         I:RK(T)         Q:40

$PROD:K(T)
         O:PK(T+1)       Q:((1-DELTA)*K0)
         O:PKT$TLAST(T)  Q:((1-DELTA)*K0)
         O:RK(T)         Q:(K0*(DELTA+R))
         I:PK(T)         Q:K0

$PROD:I(T)
         O:PK(T+1)       Q:I0
         O:PKT$TLAST(T)  Q:I0
         I:PY(T)         Q:(0.5*I0)
         I:PX(T)         Q:(0.5*I0)

$PROD:W(T) s:1
         O:PW(T)         Q:130
         I:PX(T)         Q:65
         I:PY(T)         Q:65

$DEMAND:CONS s:1
         D:PW(T)         Q:(130*QREF(T))         P:PREF(T)
         E:PL(T)         Q:(100*QREF(T))
         E:PK(TFIRST)    Q:K0
         E:PKT           Q:(-1)                  R:TK

$CONSTRAINT:TK
         SUM(T$TLAST(T), I(T)/I(T-1) - Y(T)/Y(T-1)) =G= 0;

$OFFTEXT

$SYSINCLUDE mpsgeset MOVE2_1

X.L(T) = QREF(T);
Y.L(T) = QREF(T);
W.L(T) = QREF(T);
I.L(T) = QREF(T);
K.L(T) = QREF(T);
TK.L   = K0 * (1 + G) ** CARD(T);

PX.L(T) = PREF(T);
PY.L(T) = PREF(T);
PL.L(T) = PREF(T);
PK.L(T) = (1+R)*PREF(T);
PW.L(T) = PREF(T);
RK.L(T) = PREF(T);
PKT.L   = SUM(TLAST, PK.L(TLAST)/(1+R));

MOVE2_1.ITERLIM = 0;
$INCLUDE MOVE2_1.GEN
SOLVE MOVE2_1 USING MCP;

PARAMETER REPORT1        UN REPORTE DE RESULTADOS;

          REPORT1(T, "PX") = PX.L(T);
          REPORT1(T, "PY") = PY.L(T);
          REPORT1(T, "PW") = PW.L(T);
          REPORT1(T, "PL") = PL.L(T);
          REPORT1(T, "PK") = PK.L(T);
          REPORT1(T, "RK") = RK.L(T);

          REPORT1(T, "X") = X.L(T);
          REPORT1(T, "Y") = Y.L(T);
          REPORT1(T, "W") = W.L(T);
```

```
          REPORT1(T, "I") = I.L(T);
          REPORT1(T, "K") = K.L(T);

option decimals = 4

DISPLAY REPORT1;
```

```
$TITLE move2_2.gms Modelo Dinamico Simple - Formulacion Alternativa para el
Capital

* No RK0=DELTA+R in O:RK(T), but in initial RK.L and
* in production blocks

SET      T                 /1*10/,
         TFIRST(T),
         TLAST(T);

TFIRST(T) = YES$(ORD(T) EQ 1);
TLAST(T)  = YES$(ORD(T) EQ CARD(T));


SCALAR   DELTA           FACTOR DE DESCUENTO             /0.05/
         R               TASA DE INTERES                 /0.05/
         G               TASA DE CRECIMIENTO             /0.02/
         VK              INGRESOS INICIALES DEL CAPITAL /100.0/
         K0              STOCK DE CAPITAL INICIAL
         RK0             RETORNO INICIAL DEL CAPITAL
         I0              INVERSION INICIAL;


PARAMETER
         QREF(T)         CANTIDADES
         PREF(T)         PRECIOS;

         RK0 = DELTA+R;
         K0 = VK/RK0;
         I0 = (DELTA + G) * K0;
         QREF(T) = (1 + G) ** (ORD(T) - 1);
         PREF(T) = (1/(1 + R)) ** (ORD(T) - 1);


$ONTEXT

$MODEL:MOVE2_2

$SECTORS:
         X(T)            ! Nivel de Actividad - Sector X
         Y(T)            ! Nivel de Actividad - Sector Y
         W(T)            ! Nivel de Actividad - Sector W (indice de bienestar)
         I(T)            ! Sector Inversion
         K(T)            ! Acumulacion de Capital

$COMMODITIES:
         PX(T)           ! Indice de Precios - bien X
         PY(T)           ! Indice de Precios - bien Y
         PL(T)           ! Indice de Precios - factor primario L
         PK(T)           ! Indice de Precios - factor primario K
         PW(T)           ! Indice de Precios - bienestar (funcion gasto)
         RK(T)           ! Tasa de renta del capital
         PKT             ! Restriccion del capital post terminal

$CONSUMERS:
         CONS            ! Nivel de Ingreso del consumidro CONS

$AUXILIARY:
         TK              ! Stock de Capital Terminal

$PROD:X(T) s:1
         O:PX(T)         Q:100
```

```
         I:PL(T)         Q:40
         I:RK(T)         Q:(60/RK0)      P:RK0

$PROD:Y(T) s:1
         O:PY(T)         Q:100
         I:PL(T)         Q:60
         I:RK(T)         Q:(40/RK0)      P:RK0

$PROD:K(T)
         O:PK(T+1)       Q:((1-DELTA)*K0)
         O:PKT$TLAST(T)  Q:((1-DELTA)*K0)
         O:RK(T)         Q:K0
         I:PK(T)         Q:K0

$PROD:I(T)
         O:PK(T+1)       Q:I0
         O:PKT$TLAST(T)  Q:I0
         I:PY(T)         Q:(0.5*I0)
         I:PX(T)         Q:(0.5*I0)

$PROD:W(T) s:1
         O:PW(T)         Q:130
         I:PX(T)         Q:65
         I:PY(T)         Q:65

$DEMAND:CONS s:1
         D:PW(T)         Q:(130*QREF(T)) P:PREF(T)
         E:PL(T)         Q:(100*QREF(T))
         E:PK(TFIRST)    Q:K0
         E:PKT           Q:(-1)          R:TK

$CONSTRAINT:TK
         SUM(T$TLAST(T), I(T)/I(T-1) - Y(T)/Y(T-1)) =G= 0;

$OFFTEXT

$SYSINCLUDE mpsgeset MOVE2_2

         X.L(T)  = QREF(T);
         Y.L(T)  = QREF(T);
         W.L(T)  = QREF(T);
         I.L(T)  = QREF(T);
         K.L(T)  = QREF(T);
         TK.L    = K0 * (1 + G) ** CARD(T);
         PX.L(T) = PREF(T);
         PY.L(T) = PREF(T);
         PL.L(T) = PREF(T);
         PK.L(T) = (1+R)*PREF(T);
         PW.L(T) = PREF(T);
         RK.L(T) = RK0*PREF(T);
         PKT.L   = SUM(TLAST, PK.L(TLAST)/(1+R));

MOVE2_2.ITERLIM = 0;

$INCLUDE MOVE2_2.GEN

SOLVE MOVE2_2 USING MCP;
```

```
$TITLE: move3_1.gms. Modelo Dinamico simple con inversion fuera del estado
estacionario


$ONTEXT

SAM del Modelo Estatico

            Sectores Productivos         Consumidores
Mercados|      X      Y      W          |    CONS
------------------------------------------------------
PX      |    100           -100         |
PY      |           100    -100         |
PW      |                   200         |    -200
PL      |    -40    -60                 |     100
PK      |    -60    -40                 |     100
------------------------------------------------------


SAM consistente con el estado estacionario en el periodo base

Mercados|      X      Y      W    Inv  |    CONS
------------------------------------------------------
PX      |    100            -65    -35  |
PY      |           100     -65    -35  |
PW      |                   200         |    -200
PL      |    -40    -60                 |     100
PK      |    -60    -40                 |     100
sav     |    -70     70                 |

SAM lejos del estado estacionario en el periodo base
Mercados|      X      Y      W    Inv  |     CONS
------------------------------------------------------
PX      |    100            -60    -40  |
PY      |            100    -60    -40  |
PW      |                   200         |    -200
PL      |    -40     -60                |     100
PK      |    -60     -40                |     100
sav     |                   -80     80  |

$OFFTEXT



SET      T       /1*100/,
         TFIRST(T),
         TLAST(T);

TFIRST(T) = YES$(ORD(T) EQ 1);
TLAST(T)  = YES$(ORD(T) EQ CARD(T));

SET       trep(t)          Etiquetas de los periodos para el reporte de inversion
/1*5/,
         tdecade(t)      /1,10,20,30,40,50,60,70,80,90,100/;


SCALAR   DELTA   FACTOR DE DESCUENTO                / 0.05/
         R       TASA DE INTERES                    / 0.05/
         G       TASA DE CRECIMIENTO                / 0.02/
```

```
         KZERO   INGRESOS INICIALES DEL CAPITAL     /100.00/
         K0      STOCK INICIAL DE CAPITAL
         RK0     RETORNO INICIAL DEL CAPITAL
         RKG     SUPUESTO SOBRE RK0
         I0      INVERSION INICIAL;


* EN EL CASO DE ESTADO ESTACIONARIO, EL TRABAJO Y EL CAPITAL CRECEN A LA TASA G.
* EL RETORNO INICIAL DEL CAPITAL ES RK0 = DELTA + R;
* EN EL CASO DE NO ESTADO ESTACIONARIO, EL EMPLEO CRECE A LA TASA G, EL CAPITAL
* A UNA TASA DIFERENTE, DETERMINADA POR EL RETORNO INICIAL AL CAPITAL (NOTADO
* POR RKG). NO CONOCEMOS RKG, DE MODO QUE HAREMOS UN SUPUESTO TRATANDO DE SER
* CONSISTENTES CON LA INVERSION INICIAL DADA.



PARAMETER        QREF(T) CANTIDADES
                 PREF(T) PRECIOS;

RK0 = DELTA+R;
K0  = KZERO/RK0;
I0  = (DELTA + G) * K0 + 10;

* RK0 ES DESCONOCIDA, USAMOS LAS FORMULAS PARA RK0, K0, I0 POR CONSISTENCIA
* CON EL MODELO DE EE. PODRIAMOS SIMPLEMENTE DECLARAR K0=1000; I0=80; (CONOCEMOS
* ESTOS VALORES POR LA SAM)

RKG = 0.12;


QREF(T) = (1 + G) ** (ORD(T) - 1);
PREF(T) = (1/(1+R)) ** (ORD(T) - 1);

$ONTEXT
$MODEL:DINAMICO

$SECTORS:
         X(T)    ! Nivel de actividad del sector X
         Y(T)    ! Nivel de actividad del sector Y
         W(T)    ! Nivel de actividad del sector W (indice de bienestar)
         I(T)    ! Sector Inversion
         K(T)    ! Acumulacion de Capital

$COMMODITIES:
         PX(T)   ! Indice de precios para el bien X
         PY(T)   ! Indice de precios del bien Y
         PL(T)   ! Indice de precios del factor primario L
         PK(T)   ! Indice de precios del factor primario K
         PW(T)   ! indice de precios del bienestar (funcion gasto)
         RK(T)   ! Precio de renta del capital
         PKT     ! Restriccion sobre el capital posterminal

$CONSUMERS:
         CONS    ! Nivel de ingreso del consumidor CONS

$AUXILIARY:
         TK      ! Stock de capital terminal

$PROD:X(T) s:1
         O:PX(T)        Q:100
         I:PL(T)        Q:40
         I:RK(T)        Q:(60/RKG)      P:RKG
```

```
$PROD:Y(T) s:1
        O:PY(T)         Q:100
        I:PL(T)         Q:60
        I:RK(T)         Q:(40/RKG)      P:RKG

$PROD:K(T)
        O:PK(T+1)       Q:(1-DELTA)
        O:PKT$TLAST(T)  Q:(1-DELTA)
        O:RK(T)         Q:1
        I:PK(T)         Q:1

$PROD:I(T)
        O:PK(T+1)       Q:I0
        O:PKT$TLAST(T)  Q:I0
        I:PY(T)         Q:(0.5*I0)
        I:PX(T)         Q:(0.5*I0)

$PROD:W(T) s:1
        O:PW(T)         Q:130
        I:PX(T)         Q:65
        I:PY(T)         Q:65

$DEMAND:CONS s:1
        D:PW(T)         Q:(130*QREF(T))         P:PREF(T)
        E:PL(T)         Q:(100*QREF(T))
        E:PK(TFIRST)    Q:(KZERO/RKG)
        E:PKT           Q:(-1)                  R:TK

$REPORT:
        V:C(T)          D:PW(T)         DEMAND:CONS
        V:XX(T)         O:PX(T)         PROD:X(T)
        V:YY(T)         O:PY(T)         PROD:Y(T)

$CONSTRAINT:TK
        SUM(T$TLAST(T), I(T)/I(T-1) - Y(T)/Y(T-1)) =G= 0;

$OFFTEXT

$SYSINCLUDE mpsgeset DINAMICO

X.L(T) = QREF(T);
Y.L(T) = QREF(T);
W.L(T) = QREF(T);
I.L(T) = QREF(T);
K.L(T) = K0*QREF(T);
TK.L   = K0 * (1 + G) ** CARD(T);

PX.L(T) = PREF(T);
PY.L(T) = PREF(T);
PL.L(T) = PREF(T);
PK.L(T) = (1+R)*PREF(T);
PW.L(T) = PREF(T);
RK.L(T) = RKG*PREF(T);
PKT.L   = SUM(TLAST, PK.L(TLAST)/(1+R));


SET     ITER    ITERACIONES METODO SECANTE-JACOBI  /ITER1*ITER6/;

PARAMETER
        INVEST          SENDA DE LA INVERSION
        INTRATE         TASA DE INTERES
        CONSUM          NIVEL DE CONSUMO
        GROWTH          TASA DE CRECIMIENTO DEL CONSUMO
```

```
         ERROR           ERROR ESTIMADO EN LAS ITERACIONES
         RK0VAL          VALOR DE RK0;

LOOP(ITER,
$INCLUDE DINAMICO.GEN
                 SOLVE DINAMICO USING MCP;

                 INVEST(TREP,ITER) = I.L(TREP) * I0;
                 INTRATE(TFIRST,ITER) = RKG-DELTA;
                 INTRATE(T+1,ITER)$PX.L(T+1) = PX.L(T)/PK.L(T+1) - 1;
                 CONSUM(T,ITER) = C.L(T);

*                ERROR(ITER) = C.L("1") - (200-I0);
*                ERROR(ITER) = I.L("1")*I0 - I0;

                 ERROR(ITER) = C.L("1") - (XX.L("1") + YY.L("1") - I0);
                 RK0VAL(ITER)=RKG;

                 IF(ORD(ITER)-1,
                         RKG = RKG + ERROR(ITER)/(ERROR(ITER-1)-ERROR(ITER))*
                                                 (RK0VAL(ITER-1)-RK0VAL(ITER));
                 ELSE
                         RKG = RKG + 0.01;
                   );
    );


INVEST(TREP,"I0") = I0;
DISPLAY INVEST, INTRATE, CONSUM;

$SETGLOBAL LABELS TREP
$LIBINCLUDE PLOT INVEST

CONSUM(T,ITER)$(ORD(T) GT CARD(T)-5) = NA;
GROWTH(T,ITER)$(CONSUM(T+1,ITER) NE NA)=CONSUM(T+1,ITER)/CONSUM(T,ITER)-1;
INTRATE(T,ITER)$(ORD(T) GT CARD(T)-5) = NA;

$SETGLOBAL LABELS TDECADE
$LIBINCLUDE PLOT CONSUM
$LIBINCLUDE PLOT GROWTH
$LIBINCLUDE PLOT INTRATE
```

Apéndice 4. GTAP Dinámico

El Apéndice 4 contiene la versión de Thomas Rutherford del modelo GTAP dinámico

```
$title GTAP con prevision.

SCALAR   debug    Switch para chequear la calibracion de estado estacionario /0/;

SET      t               Periodos de Tiempo (anual) /1995*2080/
         t0(t)           Primer Periodo             /1995/,
         tl(t)           Periodo Final              /2080/,
         decade(t)       Identificadores Decada     /2000, 2010, 2020,
                                                     2030, 2040, 2050,
                                                     2060, 2070, 2080 /;
PARAMETER

         year(t); year(t) = 1995 + (ord(t)-1);

SCALAR
         esubt           Elasticidad Intertemporal de Sustitucion /0.50/
         depr            Tasa de Depreciacion                     /0.07/
         gterm           Tasa de Crecimiento de largo plazo       /0.02/
         r0              Tasa de Interes en el Benchmark          /0.05/
         pk0             Precio del Capital en el Benchmark
         srvshr          Single period survival share
         termcap         Capital en el Periodo Terminal;

*        ====================================================================
*        Se calibran inversion y ahorro para un equilibrio de estado estacionario
*        a lo largo de todas las regiones

pk0 = 1 + r0;

*        ====================================================================
*        Lee la base de datos, define afiliaciones regionales y las tasas de
*        crecimiento de la linea base:

$libinclude mrtdata shock

SCALAR
         esubdm           Elasticidad de Sustitucion - domesticos vs. importaciones / 4
/,
         esubmm          Elasticidad de Sustitution - importaciones / 8 /;

PARAMETER
         gr0(r)          Tasa de crecimiento en el anho base

                         /DNK 0.02, USA 0.02, OOE 0.01,
                          ASI 0.08, JPN 0.04, CHN 0.05,
                          FSU 0.05, ROW 0.04/,

         gr(t,r)         Tasas de crecimiento a lo laro del horizonte de planeacion;

* Primera pasada: chequea consistencia de estado estacionario:

IF(debug, gr0(r) = gterm;);

PARAMETER
         qref(t,r)       Senda de referencia para las cantidades
         pref(t)         Senda de referencia para valores presentes
```

```
        a0(i,r)          Oferta - Armington
        m0(i,r)          Oferta Importada Agregada
        d0(i,r)          Oferta Domestica Importada

        c0(i,r)          Demanda Final segun Mercancia
        tc(i,r)          Tasa de Impuestos para la Demanda Final
        ct0(r)           Demanda Final Total
        pc0(i,r)         Precio de Referencia para la Demanda Final

        ld0(i,r)         Demanda de Trabajo por Sector -- Anho Base
        kd0(i,r)         Ingresos del Capital por sector -- Anho Base
        gdp0(r)          PIB -- Anho Base

        k0(r)            Acervo de CApital -- Anho Base;

*       ======================================================================
*       Se adopta alguna notacion un poco diferente de la del GTPAinGAMS:

        a0(i,r)  =  vipm(i,r)  +  vdpm(i,r)  +  vigm(i,r)  +  vdgm(i,r)  +  vifm(i,r)  +
vdfm(i,r);
        d0(i,r) = vdpm(i,r) + vdgm(i,r) + vdfm(i,r);
        m0(i,r) = vipm(i,r) + vigm(i,r) + vifm(i,r);

        c0(i,r) = vpm(i,r) + vgm(i,r);
        tc(i,r)$c0(i,r) = (tp(i,r)*vpm(i,r) + tg(i,r)*vgm(i,r)) / c0(i,r);
        ct0(r) = sum(i, c0(i,r)*(1+tc(i,r)));
        pc0(i,r) = 1 + tc(i,r);

        ld0(i,r) = vfm("LAB",i,r);
        kd0(i,r) = vfm("CAP",i,r);
        gdp0(r) = sum(i, ld0(i,r) + kd0(i,r));

PARAMETER
        vk(r)            Valor de los rendimientos de capital
        vi(r)            Valor de la inversion
        viratio(r)       Ratio de Inversion
        rk0(r)           Precio de renta del capital - benchmark
        taxrev(r)        Valor presente de los ingresos fiscales;

        vk(r) = sum(i, kd0(i,r));

*       Se asume inicialmente que la q de Tobin es 1 para todas las regiones

        rk0(r) = r0 + depr;

*       Para luego inferir el stock de capital, a partir de sus rendimientos

        k0(r) = vk(r) / rk0(r);

*        Finalmente fijamos la inversion a partir del stock de capital y de la tasa de
crecimiento:

        vi(r) = (gr0(r)+depr) * k0(r);

PARAMETER
        profit;

        profit(r,i)  =  vom(i,r)  *  (1-ty(i,r))  -  sum(j,  vafm(j,i,r)*(1+ti(j,i,r)))  -
ld0(i,r) - kd0(i,r);

DISPLAY profit;

PARAMETER
        c0adj   Incremento en el consumo para inversion,
        i0adj   Incremento en la inversion para coincidir con el benchmark
        lamda   Factor de escala para el ajuste;
```

```
         c0adj(r) = max(0, vom("cgd",r)-vi(r));
         i0adj(r) = max(0, vi(r)-vom("cgd",r));

         lamda(r) = (1-ty("cgd",r))*i0adj(r)/sum(i, c0(i,r)*pai0(i,"cgd",r));
         vafm(i,"cgd",r) = vafm(i,"cgd",r) + lamda(r) * c0(i,r);
         c0(i,r) = c0(i,r) * (1 - lamda(r));

*        La inversion supera el objetivo -- se mueve algo de la demanda de inversion
*        a la demanda final:

         lamda(r) = (1-ty("cgd",r))*c0adj(r)/ sum(i, vafm(i,"cgd",r)*pai0(i,"cgd",r));
         c0(i,r) = c0(i,r) + lamda(r) * vafm(i,"cgd",r);
         vafm(i,"cgd",r) = vafm(i,"cgd",r) * (1 - lamda(r));

         ct0(r) = sum(i, c0(i,r) * pc0(i,r));
         vom(cgd,r) = vi(r);

         profit(r,i)  =  vom(i,r)*(1-ty(i,r))-sum(j,vafm(j,i,r)*(1+ti(j,i,r)))-ld0(i,r)-
kd0(i,r);

DISPLAY  c0adj, i0adj;
DISPLAY  profit;

PARAMETER
         gdp(t,r)        Indice del PIB (relativo)
         eff(t,i,r)      Indice de Eficiencia (factor de ajuste)
         ieff(t,r)       Indice de Eficiencia en la Inversion (parametro de choque)
         keff(t,r)       Indice de Eficiencia del Capital (parametro de choque);

         ieff(t,r) = 1;
         keff(t,r) = 1;

$ontext
$model:gtap

$sectors:
         c(t,r)                  ! Consumo Privado
         y(i,t,r)$vom(i,r)       ! Producción
         a(i,t,r)$a0(i,r)             !  Agregacion  Armington  de  bienes  domesticos  e
importados
         k(t,r)                  ! Stock de Capital
         inv(t,r)                ! Inversion
         yt(t)                   ! Transporte

$commodities:
         pc(t,r)                 ! Demanda Final
         py(i,t,r)               ! Precio del producto
         pa(i,t,r)$a0(i,r)       ! Precio del bien compuesto Armington
         pl(t,r)                 ! Tasa de salario
         pk(t,r)                 ! Costo del capital
         rk(t,r)                 ! Retorno de captital
         ptk(r)                      ! Precio de una unidad de capital en el periodo
terminal
         pt(t)                   ! Servicios de Transporte
         ptax(r)                 ! Valor Presente de los ingresos fiscales

$consumers:
         ra(r)                   ! Agente Representativo
         govt(t,r)               ! Gobierno

$auxiliary:
         kt(r)                   ! Capital terminal
         theta(r)                ! Parte del stock de capital en el estado estcionario
         ta(r)                   ! Activos -- Periodo terminal

* Demanda Final
```

```
$prod:c(t,r)     s:esubt c:1
        o:pc(t,r)       q:ct0(r)
        i:pa(i,t,r)     q:c0(i,r)       p:pc0(i,r)      a:govt(t,r) t:tc(i,r)

* Produccion de combustibles no fosiles (incluye electricidad y refineria)

$prod:y(i,t,r) s:0 va:1
        o:py(i,t,r)     q:vom(i,r)      a:govt(t,r)     t:ty(i,r)
        i:pa(j,t,r)     q:vafm(j,i,r)   p:pai0(j,i,r)   a:govt(t,r)     t:ti(j,i,r)
        i:pl(t,r)       q:ld0(i,r)      va:
        i:rk(t,r)       q:kd0(i,r)      va:

$prod:inv(t,r)
        o:ptk(r)$tl(t)  q:(ieff(t,r)*vi(r))
        o:pk(t+1,r)     q:(ieff(t,r)*vi(r))
        i:py(cgd,t,r)   q:vi(r)

$prod:k(t,r)
        o:ptk(r)$tl(t)  q:((1-depr)*k0(r))
        o:pk(t+1,r)     q:((1-depr)*k0(r))
        o:rk(t,r)       q:(keff(t,r)*rk0(r)*k0(r))
        i:pk(t,r)       q:k0(r)

* Agregacion de Armington sobre domesticos e importaciones

$prod:a(i,t,r)$a0(i,r)   s:esubdm        m:esubmm        s.tl(m):0
        o:pa(i,t,r)     q:(eff(t,i,r)*a0(i,r))
        i:py(i,t,r)     q:(gdp(t,r)*d0(i,r))
        i:py(i,t,s)     q:(gdp(t,s)*vxmd(i,s,r)) p:pmx0(i,s,r) s.tl:
        + a:govt(t,s) t:tx(i,s,r) a:govt(t,r) t:(tm(i,s,r)*(1+tx(i,s,r)))
        i:pt(t)#(s)     q:(gdp(t,s)*vtwr(i,s,r)) p:pmt0(i,s,r) s.tl:
        + a:govt(t,r) t:tm(i,s,r)

* Servicios Internacionales de transporte (Cobb-Douglas):

$prod:yt(t) s:1
        o:pt(t)         q:(sum((i,r), vst(i,r)))
        i:py(i,t,r)     q:vst(i,r)

* Demanda Final:

$demand:ra(r) s:esubt
        d:pc(t,r) q:(ct0(r)*qref(t,r)) p:pref(t)
        e:ptk(r) q:(-k0(r)) r:kt(r)

* Valor de la balanza comercia en el anho base

        e:pc(t,"usa")   q:(vb(r)*qref(t,r))
        e:pl(t,r)       q:(evoa("lab",r)*qref(t,r))
        e:pl(t0,"usa")  q:(-1) r:ta(r)
        e:pk(t0,r)      q:k0(r)
        e:ptax(r)       q:taxrev(r)

$demand:govt(t,r)
        d:ptax(r)

$constraint:kt(r)
        sum(t$tl(t+1), c(t,r)*inv(t+1,r) - c(t+1,r)*inv(t,r)) =e= 0;

* Participacion de la region r en el agregado total de los activos:

$constraint:theta(r)
        theta(r) * sum((tl,rr), c(tl,rr)*pc(tl,rr)*ct0(rr)
                        - evoa("lab",rr)*qref(tl,rr)*pl(tl,rr)
                        - vb(rr)*qref(tl,rr)*pc(tl,"usa"))
                        =e= sum(tl, c(tl,r)*pc(tl,r)*ct0(r)
```

```
                                 - evoa("lab",r)*qref(tl,r)*pl(tl,r)
                                 - vb(r)*qref(tl,r)*pc(tl,"usa"));

* Ajuste para los activois en el periodo Terminal:

$constraint:ta(r)
         sum(t0, pl(t0,"usa")) * ta(r) =e=
         theta(r) * sum(rr,ptk(rr)*kt(rr)*k0(rr)) - ptk(r)*kt(r)*k0(r);

$offtext

$sysinclude mpsgeset gtap

* Tasas de crecimiento a los largo del horizonte de planeacion:

scalar ssyear /2060/;

set tss(t); tss(t) = yes$(year(t) ge ssyear);
gr(t,r)$(not tss(t)) = gterm * (year(t)-1995)/(ssyear-1995) +
                         gr0(r) * (ssyear-year(t))/(ssyear-1995);

gr(tss,r) = gterm;

*$setglobal gp_xl decade
*$libinclude gnuplot gr

qref(t,r) = 1;

LOOP(t, qref(t+1,r) = qref(t,r) * (1 + gr(t,r)););

* Genera un indice para controlar la estructura de demanda Armington:

gdp(t,r) = ( qref(t,r) * gdp0(r) / sum(rr, qref(t,rr)*gdp0(rr)) ) /
                 ( gdp0(r) / sum(rr, gdp0(rr)) );

eff(t,i,r)$a0(i,r) = ( gdp(t,r)*d0(i,r) +
                 sum(s, gdp(t,s) *
                 (vxmd(i,s,r)*pmx0(i,s,r)+ vtwr(i,s,r)*pmt0(i,s,r))) ) / a0(i,r);

DISPLAY gdp, eff;

pref(t) = (1/(1+r0))**(ord(t)-1);
pref(t) = pref(t) / sum(t0, pref(t0));

pc.l(t,r) = pref(t);
py.l(i,t,r) = pref(t);
pa.l(i,t,r) = pref(t);
pl.l(t,r) = pref(t);
pk.l(t,r) = pk0 * pref(t);
rk.l(t,r) = pref(t);
ptk.l(r) = sum(t$tl(t),pk.l(t,r))/(1+r0);
pt.l(t) = pref(t);

c.l(t,r) = qref(t,r);
y.l(i,t,r) = qref(t,r);
a.l(i,t,r) = qref(t,r);

k.l(t,r) = qref(t,r);
inv.l(t,r) = qref(t,r);
yt.l(t) = qref(t,"usa");
kt.l(r) = sum(tl, qref(tl,r))*(1 + gterm);

* Replica el banchmark calibrado de 1995 -- la desviacion debe ser
* del orden de 1e-5:

if (debug,
         gtap.iterlim = 0;
```

```
         else
         vb(r) = 0;
   );

ta.lo(r) = -inf;
theta.lo(r) = -inf;

theta.l(r) = sum(tl, c.l(tl,r)*pc.l(tl,r)*ct0(r)
                 - evoa("lab",r)*qref(tl,r)*pl.l(tl,r)
                 - vb(r)*qref(tl,r)*pc.l(tl,"usa")) /
                 sum((tl,rr), c.l(tl,rr)*pc.l(tl,rr)*ct0(rr)
                 - evoa("lab",rr)*qref(tl,rr)*pl.l(tl,rr)
                 - vb(rr)*qref(tl,rr)*pc.l(tl,"usa"));

taxrev(r) = sum(t, qref(t,r) * pref(t) *
                 (sum((i,s), gdp(t,s) * tm(i,s,r)*( vtwr(i,s,r) +
                 vxmd(i,s,r)*(1+tx(i,s,r)) )
                 + gdp(t,r) * tx(i,r,s) * vxmd(i,r,s) ) +
                 sum((i,j), ti(j,i,r)*vafm(j,i,r)) +
                 sum(i, ty(i,r) * vom(i,r)) +
                 sum(i, tc(i,r) * c0(i,r)) ));

$include gtap.gen

SOLVE gtap using mcp;

abort$debug "Benchmark precision:", gtap.objval;

parameter
         capital(t,r)  Tasa de crecimiento del capital
         consum(t,r)   Nivel de consumo
         interest(t,r) Tasa de Interes;

         capital(t,r) = 100 * (k.l(t+1,r)/k.l(t,r) - 1);
         consum(t,r) = 100 * (c.l(t+1,r)/c.l(t,r) - 1);
         interest(t,r)$(not tl(t)) = 100 * (pc.l(t,r)/pc.l(t+1,r)-1);
         capital(tl,r) = na;
         consum(tl,r) = na;
         interest(tl,r) = na;

parameter
         tasum Verificacion para los activos terminales;
         tasum = sum(r,ta.l(r)); display tasum;

display capital, consum, interest;
*.$setglobal gp_xl decade
*.$libinclude gnuplot capital
*.$libinclude gnuplot consum
*.$libinclude gnuplot interest

parameter
assets(t,*) Activos del comienzo del periodo
borrow(*,*) Endeudamiento Neto por periodo
pvdebt(*,*) Valor presente de la deuda
fvdebt(*,*) Valor futuro de la deuda;

* Deuda neta del exterior

pvdebt(t,r) = 0;
assets(t0,r) = pk.l(t0,r)*k.l(t0,r)*k0(r);
borrow(t,r) = pc.l(t,r)*c.l(t,r)*ct0(r) + py.l("cgd",t,r)*inv.l(t,r)*vi(r)
- pl.l(t,r)*evoa("lab",r)*qref(t,r)
- rk.l(t,r)*rk0(r)*k0(r)*k.l(t,r)
- govt.l(t,r);

loop(t,
         pvdebt(t+1,r) = pvdebt(t,r) + borrow(t,r);
```

```
         assets(t+1,r) = pk.l(t+1,r)*k.l(t+1,r)*k0(r) - pvdebt(t,r);
    );

fvdebt(t,r) = pvdebt(t,r)/pc.l(t,"usa");

loop(t0,
         assets(t,r) = assets(t,r)/pc.l(t0,"usa");
         borrow(t,r) = borrow(t,r)/pc.l(t0,"usa");
         pvdebt(t,r) = pvdebt(t,r)/pc.l(t0,"usa");
    );

*.$libinclude gnuplot assets
*.$libinclude gnuplot borrow
*.$libinclude gnuplot pvdebt
*.$libinclude gnuplot fvdebt

borrow(t,"chk") = sum(r, borrow(t,r));

pvdebt(t,"chk") = sum(r, pvdebt(t,r));

display assets, borrow, pvdebt, fvdebt;

borrow(t,"chk") = 0;
pvdebt(t,"chk") = 0;
```

Apéndice 5. Derivación del las FOC del problema (10)-(16).

El consumidor representativo maximiza el valor presente de su utilidad intertemporal.

$$\max \sum_{t=0}^{\infty} \left(\frac{1}{1+\rho} \right)^{t} U(C_t)$$

Las restricciones son: $C_t \leq F(K_t, L_t) - I_t$, $K_{t+1} = K_t (1-\delta) + I_t$.

Considérense dos períodos: t y t+1. El agente representativo toma decisiones no solamente acerca de su consumo hoy, sino acerca de su consumo hoy y de la inversión de mañana (que determinará su capital y consumo mañana). Así, se tienen tres variables de decisión: c(t), k(t), and i(t).

Nota: Esto supone considerar dos restricciones sobre el capital, una para K_t y otra para K_{t+1}.

El Lagrangeano es en consecuencia:

$$L = (1/(1+\rho))^{t}\, U(C_t) + \lambda_1 (F(K_t, L_t) - I_t - C_t) + \lambda_2 (K_{t-1} (1-\delta) + I_{t-1} - K_t) + \lambda_3 (K_t (1-\delta) + I_t - K_{t+1})$$

Se tienen tres multiplicadores, λ_1 es el de la restricción sobre el consume, λ_2 es el de la restricción de capital hoy, y λ_3 es el de la restricción sobre el capital de mañana.

FOC:

$$\partial L/\partial C_t = (1/(1+\rho))^{t} (\partial U(C_t)/\partial C_t) - \lambda_1 = 0$$

$$\partial L/\partial K_t = \lambda_1 (\partial F/\partial K_t) - \lambda_2 + \lambda_3 (1-\delta) = 0$$

$$\partial L/\partial I_t = -\lambda_1 + \lambda_3 = 0$$

Que son las mismas ecuaciones (14)–(16) del documento. Los multiplicadores pueden ser entendidos así: $\lambda_1 = P_t$, $\lambda_2 = PK_t$, $\lambda_3 = PK_{t+1}$.